Mở đầu

Maple 6 là phần mềm toàn diện để giải quyết các bài toán cao cấp. Bao gồm những công cụ xử lý, tính toán trong các lãnh vực toán học như :

1. Đại số tuyến tính: Ma trận, Định thức, Hệ phương trình tuyến tính, Không gian véctơ …
2. Giải tích: Hàm số, giới hạn, Liên tục , Đạo hàm, Tích phân, Phương trình vi phân, Chuỗi …
3. Đồ họa, Toán rời rạc, Thống Kê, … và nhiều lãnh vực khác của toán học.

Với trên 3000 hàm số Maple là một trợ lý toán học tuyệt vời giúp giải quyết phần tính toán trong học tập và nghiên cứu.

Maple làm việc theo câu lệnh nhập từ bàn phím và có thể lưu thành tập tin để sử dụng lại khi cần.

Một số điều qui định khi nhập lệnh:

1. Kết thúc câu lệnh : Mỗi câu lệnh được kết thúc bởi dấu **;** (thì in kết quả ra màn hình) hoặc dấu **:** (không in kết quả)
2. Thi hành câu lệnh : Sau khi kết thúc lệnh thì ấn phím Enter để thực hiện lệnh.
3. Các câu lệnh có thể được đánh dấu, sao chép theo cách thức như trong hệ điều hành Windows

Một số điều cần chú ý:

1. Có phân biệt chữ hoa và chữ thường.
   Ví dụ: Int và int là hai lệnh khác nhau
2. Để tạo một chú thích cho câu lệnh, ta dùng dấu # trước đoạn văn ghi chú.
   Ví dụ: # Tính tích phân
3. Dùng lệnh **restart** để khởi tạo mới các biến, hàm đã sử dụng trước đó.
4. Cần tra cứu cú pháp câu lệnh ta dùng mục Help trên thanh thực đơn của Maple. Muốn tra cứu nhanh thì dùng dấu ? và tên mục cần tra cứu.
   Ví dụ: ?plot
   ?ifactor

Dữ liệu trong Maple

1. Các phép toán:

a. Số học : +, - , * , / , ^ hay ** , !

Trong Toán	Trong Maple
a+b	a+b
a-b	a-b
a.b	a*b
a/b	a/b
A^n	A^n
A^n	A**n
A!	A!

Ví dụ : Tính biểu thức A= $\dfrac{2^2+5}{2(\sqrt{2}+1)}$

```
> A=(2^2+5)/(2*(2^(1/2)+1));
```

b. So sánh: < , <= , > , >= , = , <>

Trong Toán	Trong Maple
x=2	x=2
$x\neq4$	x<>4
$1\leq x$	1<=x
3<x	3<x
x>a	x>a
$t\geq3$	t>=3

c. Logic : and , or , not

Ví dụ :

Trong Toán	Trong Maple
$0\leq x\leq3$	(0<=x)and(x<=3)
x<1 ; $2\leq x$	(x<1) or (2<=x)
$x\notin[0,1]$	not ((0<=x)and(x<=3))

Ví dụ :

```
> evalb(5>4 and 7<1);
```

2. **Kiểu dữ liệu**:

Trong Maple ta có các kiểu dữ liệu sau đây :

Dữ liệu	**Tên Kiểu**	**Ví dụ**
Số nguyên	integer	123
Phân số	fraction	12/3
Số thực	float	12.3456
Số phức	complex	a+bI (I chữ hoa)
Xâu kí tự	string	"ab cd12"
Dãy	exprseq	a,b,c
Tập hợp	set	{a,b,c}
Danh sách	list	[a,b,c]
Miền	range	1..3 hay a..f

3. **Hằng**: là các giá trị cài sẵn của Maple có giá trị không đổi. Một số hằng của Maple được liệt kê dưới đây:

Hằng	**Tên hằng**	**Giá trị**
π	Pi	3.141592654
e	E	2.718282828
Catalan $C=\sum_{n=1}^{\infty}\frac{(-1)^n}{(2n+1)^2}$	Catalan	0.915955942
$\gamma=\lim_{n\to\infty}\left(\sum_{k=1}^{n}\frac{1}{k}\right)-\ln n$	gamma	0.5772156649
∞	infinity	
Đúng , sai	true , false	

a) Để xem danh sách tên các hằng dùng lệnh : constants;

b) Thêm tên hằng vào danh sách hằng :

constants:=constants, tên_hằng_mới ;

c) Gán giá trị cho hằng mới: **macro**(tên_hằng_mới = giátrị) ;

<u>Ví dụ</u> : Thêm hằng K=9.109558. 10^{-31}

constants:= constants K;

macro(K=9.109558*10^(-31));

4. **Biến** là vùng nhớ lưu giá trị và được truy xuất qua tên của biến.

- **<u>Tên biến</u>**: là tên gọn của biến gồm các chữ cái (a..z , A..Z) , các chữ số (0..9) và dấu gạch nối "_" . Tên biến phải bắt dầu chữ cái hoặc dấu "_" và có phân biệt theo chữ in và chữ thường.
  Tên biến không được trùng với các từ khóa dành riêng của Maple gồm:
  and break by catch description do done elif else end error export fi finally for from global if in intersect local minus mod module next not od option options or proc quit read return save stop then to try union use while
- **<u>Kiểu biến</u>**: là một trong các kiểu dữ liệu ở phần 2)

5. **Biểu thức**: thực hiện một số hữu hạn các phép toán trên các biến, hằng và hàm số phù hợp kiểu dữ liệu.

<u>Ví dụ</u> : A = $\frac{\sqrt[3]{tgx}+1}{\sin^2 x+1}$

```
A:=(tan(x)^(1/3)+1)/(sin(x)^2+1);
```

Ví dụ: $B = \dfrac{2^{\sin x} + \ln(2x-3)}{\cos\left(\dfrac{\pi}{x}\right) - \sqrt[3]{x}}$

```
B:=(2^sin(x)+ln(2*x-3))/(cos(Pi/x)-x^(1/3));
```

6. **Phép gán** : Để đưa giá trị vào vùng nhớ ta dùng phép gán (:=) như sau :

Tên_biến := ***biểu _thức_giá_trị*** **;**

Ví dụ

```
a:= 12 ;
b:= 2^3+1/2 ;
c:=(2+Pi) /( E + Catalan);
```

ĐẠI SỐ

I. BIỂU THỨC ĐẠI SỐ

1) Số nguyên

a) Số nguyên tố :

Các hàm số liên quan đến số nguyên tố

Tên hàm số	ý nghĩa
isprime(n)	kiểm tra số n có là số nguyên tố không.
nextprime(n)	số nguyên tố nhỏ nhất và ≥ n
prevprime(n)	số nguyên tố lớn nhất và ≤ n
ithprime(n)	số nguyên tố thứ n
ifactor(n)	thừa số nguyên tố của n

Ví dụ:

isprime(113); -> true
nextprime(90); -> 97
prevprime(90); -> 89
ithprime(7); -> 17 ‘số nguyên tố thứ 7

Ví dụ: Phân tích 2004 thành các thừa số nguyên tố.

`ifactor(2004);` -> $(2)^2\ (3)\ (167)$

b) Ước số chung – Bội số chung:

igcd(n_1,n_2,…)	ước số chung lớn nhất của n_1,n_2,…
ilcm(n_1,n_2,…)	bội số chung nhỏ nhất của n_1,n_2,…

Ví dụ:

`igcd(24,16,112);` -> 8
`ilcm(8,12,9);` -> 72

2) Khai triển:

Lệnh **expand(Bthức)** sẽ khai triển biểu thức đại số theo các qui tắc lũy thừa,hàm mũ, hàm logarit, lượng giác.

Ví dụ:

`expand((x^2+1)*(x+a)/x);` -> $x^2 + x\,a + 1 + \dfrac{a}{x}$

`expand((x^2+1)^3);` -> $x^6 + 3\,x^4 + 3\,x^2 + 1$

`expand(sin(x+y));` -> $\sin(x)\cos(y) + \cos(x)\sin(y)$

`expand(exp(x-y));` -> $\frac{e^x}{e^y}$

3) Rút gọn biểu thức số:

- Lệnh **combine(Bthức, name)** với name : power,exp, trig…
Kết hợp các số hạng của biểu thức đại số theo các công thức lũy thừa, hàm mũ, hàm logarit, lượng giác… ngược lại lệnh **expand.**
Ví dụ :

`combine(exp(x)^2*exp(y),exp);` -> $e^{(2x+y)}$

`combine((x^a)^2,power);` -> $x^{(2a)}$

`combine(2*sin(x)*cos(x),trig);` -> $\sin(2x)$

- Lệnh **simplify(Bthức)** đơn giản rút gọn biểu thức đại số theo các qui tắc lũy thừa, mũ, logarit, lượng giác.
Ví dụ: `simplify(4^(1/2)+3);` -> 5

Ví dụ: Rút gọn $2\cos^3 x + \sin x \sin(2x)$
`simplify(2*cos(x)^3+sin(x)*sin(2*x));`
-> $2\cos(x)$

Ta có thể dùng `simplify(bthuc,dk)` *tính giá trị biểu thức với hệ điều kiện ràng buộc của các biến trong* `bthuc` .

Ví dụ : Gọi a, b là hai nghiệm của phương trình:
$x^2 - x - 3 = 0$
Tính giá trị biều thức B= $a^2 + b^2$

Theo định lý viét a và b có ràng buộc : a+b =1 , ab = -3 nên :
`B=simplify(a^2+b^2,{a+b=1,a*b=-3});`
-> $B = 7$

Ví dụ: Cho a, b, c là số thực thỏa :
$a+b+c = 3$, $a^2+b^2+c^2 = 9$, $a^3+b^3+c^3 = 24$.
Tính A = $a^4+ b^4 +c^4$
`dk:={a+b+c=3,a^2+b^2+c^2=9,a^3+b^3+c^3=24};`
$dk := \{a+b+c=3,\ a^2+b^2+c^2=9,\ a^3+b^3+c^3=24\}$
`A=simplify(a^4+b^4+c^4,dk);` $A = 69$

- Lệnh **collect** nhóm các số hạng theo các biến hoặc hàm.

Ví dụ: nhóm x(x+1) + y(x+1) theo x
`collect(x*(x+1)+y*(x+1),x);` -> $x^2 + (1+y)x + y$

Ví dụ: nhóm `alnx-xlnx-x` theo hàm lnx
`collect(a*ln(x)-ln(x)*x-x,ln(x));`
$(a-x)\ln(x) - x$

Ví dụ: Cho f= $a^3x - x + a^3 + a$
`f := a^3*x-x+a^3+a;` -> $f := a^3 x - x + a^3 + a$
nhóm theo x : `collect(f,x);` -> $(a^3-1)x + a^3 + a$
nhóm theo x và thừa số hệ số của x:
`collect(f,x,factor);`
-> $(a-1)(a^2+a+1)x + a(a^2+1)$

Ví dụ: Cho g = xy + axy + yx^2 – ayx^2 + x + ax

```
g :=x*y+a*x*y+y*x^2-a*y*x^2+x+a*x;
```

-> $g := x\,y + a\,x\,y + y\,x^2 - a\,y\,x^2 + x + a\,x$

Nhóm g theo x: `collect(g,x);`

-> $(y - a\,y)\,x^2 + (y + a\,y + 1 + a)\,x$

Nhóm g theo x và hệ số theo y : > `collect(g,[x,y]);`

-> $y\,(1 - a)\,x^2 + ((1 + a)\,y + 1 + a)\,x$

4) **Tính giá trị**: dùng lệnh **evalf(Bthức, n)** , n số chữ số

Ví dụ: `cos(1) + sin(1);` -> $\cos(1) + \sin(1)$

`evalf(cos(1) + sin(1));` -> 1.381773291

`evalf(cos(1) + sin(1),7);` -> 1.381773

5) **Đổi dạng số** : Lệnh **convert(bthức , 'kiểu')**

với kiểu là : int, float, binary, hex, fraction …

`convert(1215,'hex');` -> **4BF**

`convert(1215,'binary');` -> 10010111111

`convert( 1.23456, fraction );` -> $\frac{3858}{3125}$

II. ĐA THỨC

1) Phép toán đa thức : +, -, *, /, ^

Ví dụ :

`f:=(x-2)*(x+1)^2;` -> $f := (x - 2)\,(x + 1)^2$

`g:=x-1;` -> $g := x - 1$

`f/g;` -> $\frac{(x - 2)\,(x + 1)^2}{x - 1}$

- Thương số trong phép chia đa thức f/g biến x là : **quo(f,g,x);**

  `quo(f,g,x);` -> $x^2 + x - 2$

- Dư số trong phép chia đa thức f/g biến x là : **rem(f,g,x);**

  `rem(f,g,x);` -> -4

  `rem((x-2)*(x+1)^2,x-1,x);` -> -4

- Hệ số đa thức :

  `coeffs(2*x^3-3*x+1,x);` -> **1,-3,2**

- Bậc đa thức

  `degree((2*x^3+1)*(1-x^2),x);` -> **5**

- Ước số chung lớn nhất: **gcd(f,g);**

  `gcd(x^2-3*x+2,x^2-4);` -> **x-2**

2) Nghiệm đa thức:

- Lệnh **roots(f)** cho ra *nghiệm hữu tỷ* dạng:

  [$[x_1,n_1]$ … $[x_k, n_k]$]

  với kí hiệu $[x_1,n_1]$ nghĩa là nghiệm x_1, bội n_1 : $(x - x_1)^{n_1}$

  Ví dụ :

  `roots(x^3-3*x^2+4);` -> [[2, 2], [-1, 1]]

  `roots(x^4-4);` -> **[]** (không có nghiệm hữu tỷ)

- Lệnh **solve(f , x)** cho ra nghiệm thực hoặc nghiệm phức:

  `solve(x^4-4,x);` -> $I\sqrt{2}, -I\sqrt{2}, \sqrt{2}, -\sqrt{2}$

3) Phân tích đa thức thành tích số:

- Lệnh **factor(f);**

  `factor(x^4-4);` -> $(x^2-2)(x^2+2)$

- Lệnh **factor(f,real);** hoặc **factor(f,complex);**

  `factor(x^4-4,real);`

  -> $(x+1.414213562)(x-1.414213562)(x^2+1.999999999)$

  `factor(x^4-4,sqrt(2));`

  -> $(x^2+2)(x-\sqrt{2})(x+\sqrt{2})$

III. HÀM HỮU TỶ

1. **Phép tính** :

Tên lệnh	ý nghĩa
numer(f)	Tử số của biểu thức hữu tỷ f
denom(f)	Mẫu số của biểu thức hữu tỷ f
normal(f)	Tối giản biểu thức hữu tỷ f

Ví dụ : `f:=(((x-2)^3)/(x^2-4))+x/(x-1);`

$$f := \frac{(x-2)^3}{x^2-4} + \frac{x}{x-1}$$

`numer(f);`

$$x^4 - 6x^3 + 18x^2 - 24x + 8$$

`denom(f);`

$$(x^2-4)(x-1)$$

`normal(f);`

$$\frac{x^3 - 4x^2 + 10x - 4}{(x-1)(x+2)}$$

2. **Khai triển phân thức thành tổng phân thức đơn giản:**

Lệnh: **convert(f ,parfrac , x);**

(parfrac = partial fraction form)

Ví dụ:

`f:=(((x-2)^3)/(x^2-4))+x/(x-1);`

$$f := \frac{(x-2)^3}{x^2-4} + \frac{x}{x-1}$$

`convert(f,parfrac,x);`

$$x - 5 + \frac{16}{x+2} + \frac{1}{x-1}$$

IV. GIẢI PHƯƠNG TRÌNH – HỆ PHƯƠNG TRÌNH

1) Giải phương trình, bất phương trình:

Dùng lệnh: **solve(eqn, var)**

trong đó **eqn** là phương trình hoặc bất phương trình ẩn x

Ví du 1 : Giải phương trình : x^4-5x^2 + 6x = 2

```
solve(x^4-5*x^2+6*x=2,x);
```

-> $1, 1, -1+\sqrt{3}, -1-\sqrt{3}$

Ví dụ 2: Giải phương trình x^2 -2ax =1

```
epn:=x^2-2*a*x=1;
```

-> $epn := x^2 - 2\,a\,x = 1$

```
solve(epn,x);
```

-> $a+\sqrt{a^2+1},\ a-\sqrt{a^2+1}$

Ví dụ 3: Giải bất phương trình : x^2 + 2x -4 > 0

```
solve( x^2+2*x-4 >0, x );
```

$\mathrm{RealRange}(-\infty, \mathrm{Open}(-1-\sqrt{5})), \mathrm{RealRange}(\mathrm{Open}(-1+\sqrt{5}), \infty)$

Ví dụ 4: Giải bất phương trình : $\sqrt{x-2}+\sqrt{4-x} \le 2$

```
eqn:=sqrt(x-2)+sqrt(4-x)<=2:
solve(eqn,x);
```

-> $\mathrm{RealRange}(2, 4)$

Chú ý : Ta có thể gán nghiệm vào biến, để đánh giá gần đúng các nghiệm như sau :

Ví dụ 5: Giải phương trình: x^4 - 5x^2 + 6x = 2

```
sols := [solve(x^4-5*x^2+6*x=2,x)];
```

$sols := [1, 1, -1+\sqrt{3}, -1-\sqrt{3}]$

```
evalf(sols);
```

-> [1., 1., .732050808, -2.732050808]

Ví dụ 6: Giải phương trình: x^4 + x + 1 = 0

```
solve(x^4+x+1,x);
```

$\mathrm{RootOf}(\_Z^4+\_Z+1, index=1), \mathrm{RootOf}(\_Z^4+\_Z+1, index=2),$
$\mathrm{RootOf}(\_Z^4+\_Z+1, index=3), \mathrm{RootOf}(\_Z^4+\_Z+1, index=4)$

có nghiệm phức tính gần đúng bởi lệnh :

```
evalf({%});
```

$\{\{-.7271360845+.4300142883\,I, -.7271360845-.4300142883\,I,$
$.7271360845-.9340992895\,I, .7271360845+.9340992895\,I\}\}$

Để giải phương trình đệ qui ta dùng lệnh **rsolve** như ví dụ sau :

Ví dụ: Cho dãy số Fibonacci f(0) = 0, f(1) = 1 , f(n+1) = f(n+1) + f(n).
Tìm f(n)

```
rsolve({f(n+2)=f(n+1)+f(n),f(0)=0,f(1)=1},f(n));
```

$$\frac{\left(1-\frac{1}{5}\sqrt{5}\right)\left(2\,\frac{1}{-1+\sqrt{5}}\right)^n}{-1+\sqrt{5}}+\frac{\left(-\frac{1}{5}\sqrt{5}-1\right)\left(-2\,\frac{1}{1+\sqrt{5}}\right)^n}{1+\sqrt{5}}$$

2) Giải hệ phương trình, hệ bất phương trình:

Ví dụ 1: Giải hệ phương trình: $\begin{cases} x^2+y^2=25 \\ x-y=7 \end{cases}$

```
solve({x^2+y^2=25, x-y=7});
```

$\{x=3, y=-4\}, \{x=4, y=-3\}$

Ví dụ 2: Giải và biện luận hệ phương trình : $\begin{cases} mx+y+z=1 \\ x+my+z=1 \\ x+y+mz=1 \end{cases}$

```
hpt:={m*x+y+z=1,x+m*y+z=1,x+y+m*z=1};
```

$$hpt := \{m\,x+y+z=1,\ x+m\,y+z=1,\ x+y+m\,z=1\}$$

```
solve(hpt);
```

$$\{m=1, z=z, x=-z+1-y, y=y\}, \{y=z, m=-\frac{-1+2z}{z}, z=z, x=z\}$$

Ví dụ 3: Giải hệ phương trình : $\begin{cases} u+v+w\ =1 \\ 3u+v\ \ \ \ =3 \\ u-2v-w=0 \end{cases}$

```
hpt := {u+v+w=1, 3*u+v=3, u-2*v-w=0};
```

$$hpt := \{u-2\,v-w=0,\ 3\,u+v=3,\ u+v+w=1\}$$

```
solve(hpt);
```
-> $\{w=\frac{-2}{5}, v=\frac{3}{5}, u=\frac{4}{5}\}$

Ví dụ 4: Giải hệ bất phương trình : $\begin{cases} x^2-3x+2\geq 0 \\ 4-x^2\geq 0 \end{cases}$

```
bpt:={x^2-3*x+2>=0,4-x^2>=0};
```

$$bpt := \{0\leq 4-x^2,\ 0\leq x^2-3\,x+2\}$$

```
solve(bpt);
```
-> $\{-2\leq x, x\leq 1\}, \{x=2\}$

3) <u>Giải gần đúng</u> : phương trình hoặc bất phương trình ta dùng lệnh: **fsolve(eqns, vars, options);**

- eqns là phương trình hoặc hệ phương trình.
- vars là tập hợp ẩn.
- options là tham số điều khiển lời giải như: complex, a..b, …

Ví dụ 1: Giải phương trình : tg(sinx)=1

```
solve(tan(sin(x))=1,x);
```
-> $\arcsin\left(\frac{1}{4}\pi\right)$

```
fsolve( tan(sin(x))=1, x );
```
-> .9033391108

Ví dụ 2: Tìm nghiệm phương trình : $23x^5 + 105x^4 - 10x^2 + 17x = 0$ thỏa $x\in[-1,1]$

```
poly := 23*x^5 + 105*x^4 - 10*x^2 + 17*x:
fsolve( poly, x, -1..1 );
```

-.6371813185, 0.

Ví dụ 3: Giải hệ phương trình :

$\begin{cases} \sin(x+y)-e^x y=0 \\ x^2-y=2 \end{cases}$ thỏa $x\in[-1,1]$, $y\in[-2,0]$

```
f := sin(x+y) - exp(x)*y = 0:
g := x^2 - y = 2:
fsolve({f,g},{x,y},{x=-1..1,y=-2..0});
```

$\{x=-.6687012050, y=-1.552838698\}$

Thực hành

1. Tìm ước số chung lớn nhất của 1242 và 1024
2. Phân tích thừa số nguyên tố của N và suy ra số ước số của N

   N= 9876543210123456789
3. Tính giá trị biểu thức với chín số lẻ

   $$\sqrt{2\sqrt{19549}+286}+\sqrt[3]{28}-e^{\pi}$$
4. Chứng minh rằng : $\sqrt{2\sqrt{19549}+286}$ bằng $\sqrt{113}+\sqrt{173}$
5. Rút gọn các biểu thức :
   a) $\sqrt{13+30\sqrt{2+\sqrt{9+4\sqrt{2}}}}$
   b) $\dfrac{2+\sqrt{3}}{\sqrt{2}+\sqrt{2+\sqrt{3}}}+\dfrac{2-\sqrt{3}}{\sqrt{2}-\sqrt{2-\sqrt{3}}}$
6. Cho biểu thức : $A=(x^2+xy+x+y)(x+y)$. Hãy biến đổi biểu thức A về dạng:
   a) $x^3+2x^2y+xy^2+x^2+2xy+y^2$
   b) $(x+1)(x+y)^2$
   c) $y^2+(2y+y^2)x+(1+2y)x^2+x^3$
   d) $x^3+x^2+(2x^2+2x)y+(x+1)y^2$
7. Cho phân thức $f=\dfrac{x^4+x^3-4x^2-4x}{x^4+x^3-x^2-x}$, biến đổi f về dạng :
   a) $\dfrac{x^2-4}{x^2-1}$
   b) $\dfrac{(x-2)(x+2)}{x^2-1}$
8. Tìm miền xác định của hàm số :
   a) $f(x) = \lg(\sqrt{x-4}+\sqrt{6-x})$
   b) $g(x) = \arcsin\left(\dfrac{x^2-x}{x^2-1}\right)$
9. Tìm đa thức bậc 2 đi qua 3 điểm: (-2;36) , (1;120) , (-3;48)
10. Tìm F(n) thỏa : $F(n) = F(n-1) + n^2$, $F(1) = 0$
11. Tìm p , q , r , s , t để phản ứng sau cân bằng.

    $$pCO+qCO_2+rH_2\rightarrow sCH_4+tH_2O$$
12. Giải phương trình :

    $48x^5 + 8x^4 - 6x^3 + 114x^2 - 37x + 18 = 0$

HÀM SỐ & ĐỒ THỊ

1. HÀM SỐ CƠ BẢN

Maple định nghĩa các hàm số dùng cho từng kiểu dữ liệu như :

a) Các hàm số cho số nguyên (Integer)

Tên hàm số	ý nghĩa
abs(x)	trị tuyệt đối của x
min(x_1, x_2, …)	giá trị nhỏ nhất của x_1, x_2,…
max(x_1, x_2, …)	giá trị lớn nhất của x_1, x_2,…
irem(m,n)	dư số trong phép chia m/n
iquo(m,n)	thương số trong phép chia m/n
igcd(n_1,n_2,…)	ước số chung lớn nhất của n_1,n_2,…
ilcm(n_1,n_2,…)	Bội số chung nhỏ nhất của n_1,n_2,…
isprime(n)	kiểm tra xem số n có là số nguyên tố không.
nextprime(n)	số nguyên tố nhỏ nhất và $\geq$ n
prevprime(n)	số nguyên tố lớn nhất và $\leq$ n
ithprime(n)	số nguyên tố thứ n trong dãy các số nguyên tố
ifactor(n)	thừa số nguyên tố của n

Ví dụ: irem(23,4) ; -> 3 iquo(23,4); -> 5
igcd(24,16,112); -> 8 ilcm(8,12,9); -> 72

b) Các hàm số cho số thực (Float)

Tên hàm số	ý nghĩa
exp(x)	e^x
ln(x) hay log(x)	logarit nêpe (cơ số e) của x
log10(x) , log[b](x)	logarit thập phân lgx, logarit cơ số b
sqrt(x)	hàm căn bậc hai : $\sqrt{x}$
sin(x) , cos(x) , tan(x), cot(x)	sinx , cosx, tgx, cotgx
sec(x) , csc(x)	1/cosx , 1/sinx
arcsin(x) , arccos(x), arctan(x), arccot(x)	arcsinx , arccosx, arctgx , arccotgx
sinh(x) , cosh(x)	$\dfrac{e^x - e^{-x}}{2}$, $\dfrac{e^x + e^{-x}}{2}$
tanh(x) = sinh(x) / cosh(x)	$\dfrac{e^x - e^{-x}}{e^x + e^{-x}}$
coth(x) = cosh(x) / sinh(x)	$\dfrac{e^x + e^{-x}}{e^x - e^{-x}}$

Ví dụ: Biểu diễn biểu thức : A= $\sqrt[3]{\sin x} + \dfrac{\lg x + 2^x}{tgx}$ theo Maple

```
A:=sin(x)^(1/3)+(log10(x)+2^x)/tan(x);
```

Ví dụ: Biểu diễn thành lũy thừa hàm sinh(x) và hàm sec(x) theo sinx, cosx

```
convert(sinh(x),exp) ;
convert(sec(x),sincos) ;
```

c) Các hàm số cho số phức (complex)

Tên hàm số	ý nghĩa
Re(z)	Phần thực của số phức z
Im(z)	Phần ảo của số phức z
argument(z)	Argument của số phức z
abs(z)	|z| môđun của số phức z

Ví dụ: `Im(exp(I));` -> sin(1)
`argument(-1);` -> π
`argument(3+4*I);` -> arctg(4/3)
`abs(3+4*I);` -> 5

Ví dụ: Một số hàm số thực cũng dùng được cho số phức
`sqrt(-4);` -> **2I**
`sqrt(3+4*I);` -> **2+I**
`sqrt(2.0);` -> 1.414213562
`sqrt(2)` -> $\sqrt{2}$

2. ĐỊNH NGHĨA HÀM SỐ MỚI

Ngoài các hàm số đã được định nghĩa sẵn, Maple cung cấp cho người sử dụng công cụ để tạo thêm các hàm số mới theo cú pháp sau:

TênHàmSố := (DanhSáchBiến) -> CôngThứcHàmSố ;

Ví dụ 1: Hãy định nghĩa hàm số f(x) = $\dfrac{x^2+1}{\sqrt{x}+1}$

```
f:= x-> (x^2+1)/(sqrt(x)+1);
```

Tính giá trị hàm số tại x=2 dùng lệnh : `f(2);` -> $5\dfrac{1}{\sqrt{2}+1}$

Ví dụ 2: Định nghĩa hàm hai biến số g(x,y)=$\sqrt[3]{x^2+y^2+xy}$

```
g:=(x,y)->(x^2+y^2+x*y)^(1/3);
```

$g := (x, y) \to (x^2+y^2+x\,y)^{(1/3)}$

`g(1,2),g(-1,2);` $7^{(1/3)}, 3^{(1/3)}$

- Trường hợp hàm xác định bởi nhiều công thức, ta dùng lệnh :

piecewise(cond_1 , f_1 , cond_2 , f_2, . . . , cond_n, f_n , f_otherwise) ;

Ví dụ 3: Định nghĩa hàm số f(x) = $\begin{cases} x & \text{khi } x < 1 \\ 0 & \text{khi } x = 1 \\ e^x & \text{khi } x > 1 \end{cases}$

```
f:=x->piecewise(x<1, x, x=1 , 0, exp(x));
```

$f := x \to \text{piecewise}(x < 1, x, x = 1, 0, e^x)$

`f(2);` -> e^2

- **Hàm đệ qui** :

* **If** *ĐiềuKiện* **then** *Côngthức1* **else** *CôngThức2* **fi;**

hoặc

* **If** *ĐiềuKiện1* **then** *Côngthức1*
elif *ĐiềuKiện2* **then** *CôngThức2*
else *CôngThức3* **fi;**

trong đó các từ in đậm là từ khóa bắt buộc phải có.

Ví dụ 4 : Định nghĩa dãy số Lucas L_n bởi công thức :

$L_1=1$, $L_2=3$ và $L_n = L_{n-1} + L_{n-2}$

Nhập lệnh như dưới đây và dùng *Shift+Enter* để xuống dòng:

```
L:=n->
 if not type(n,'nonnegint')
 then ERROR("n la so nguyen duong ")
 elif n=1 then 1
 elif n=2 then 3
 else L(n-1)+L(n-2)
 fi;
```

```
L(3);  -> 4
L(4);  -> 7
```

3. CÁC PHÉP TOÁN TRÊN HÀM SỐ

Phép Toán	ý nghĩa
f+g	Phép cộng
f-g	Phép trừ
f*g	Phép nhân
f/g	Phép chia
f@g	Phép hợp
f@@n	Phép hợp f^n

Ví dụ :

* `f:=x->(x^2-1)/(x-1);` -> $f := x \to \frac{x^2-1}{x-1}$

`g:=f+ln;` -> $g := f + \ln$

`g(x);` -> $\frac{x^2-1}{x-1} + \ln(x)$

* `h:=g@f;` -> $h := (f + \ln)@f$

`h(x);`

$$\frac{\frac{(x^2-1)^2}{(x-1)^2} - 1}{\frac{x^2-1}{x-1} - 1} + \ln\left(\frac{x^2-1}{x-1}\right)$$

* `F:=f@@2;`

-> $F := f^{(2)}$

`F(x);`

$$\frac{\frac{(x^2-1)^2}{(x-1)^2} - 1}{\frac{x^2-1}{x-1} - 1}$$

4. VẼ ĐỒ THỊ HÀM SỐ

a. <u>Hàm 1 biến</u> : dùng lệnh **flot(f, h, v, option1, option2,…);**

trong đó:

- f : hàm số thực hoặc biểu thức chứa x
- h : miền ngang (horizontal range) dạng a..b hoặc x=a..b
- v : miền dọc (vertical range) tùy chọn
- option gồm :
    - tilte =" Tiêu đề đồ thị "
    - titlefont= [family, style, size]
    - color=n
    - style = point , line, patch …
    - numpoints : số điểm vẽ (độ mịn)
    - axes = none, normal, boxed, framed
    - legend=[danh sách chú thích]

<u>Ví dụ</u> : Vẽ đồ thị hàm số $y=\dfrac{x}{x^2+1}$

```
plot(x/(x^2+1),x=-10..10,title="Do thi ham so");
```



<u>Ví dụ 2</u> : Vẽ đồ thị hàm số y=sinx (màu đỏ, từng điểm) và $y=x-\dfrac{x^3}{6}$ (màu xanh, kiểu line)

```
plot([sin(x), x-x^3/6], x=-2..2,
      color=[red,blue],
       style=[point,line]);
```



Ví dụ 3 : Vẽ đồ thị hàm số y= $\begin{cases} x^2+1 & \text{khi } x<1 \\ 3-x & \text{khi } x \ge 1 \end{cases}$

Khi viết lệnh plot không dùng x = -3..3 mà dùng -3..3

```
f:=x-> if x<1 then x^2+1 else 3-x fi;
plot(f,-3..3);
```



b) Hàm dạng tham số:

- Hệ tọa độ Descartes: **plot([x(t), y(t),t=a..b], option);**

Ví dụ 1 : Vẽ đồ thị đường astroid $\begin{cases} x = 2\cos^3 t \\ y = 2\sin^3 t \end{cases}$

```
plot([2*cos(t)^3,2*sin(t)^3,t=0..2*Pi]);
```



Ví dụ 2: Vẽ đồ thị hai hàm số $\begin{cases} x = 2\cos^3 t \\ y = 2\sin^3 t \end{cases}$; $\begin{cases} x = t \\ y = t^2 \end{cases}$ trên cùng một hệ trục tọa độ.

```
plot({[2*cos(t)^3,2*sin(t)^3,t=0..2*Pi],
       [t,t^2,t=-2..2]});
```



- **<u>Hệ tọa độ cực:</u>**

  plot([r(t) , φ(t), t=t_0 .. t_1], coords=polar,option);

<u>Ví dụ 1:</u> Vẽ đồ thị : r=$\frac{\sin(\varphi)}{\varphi}$

```
plot([sin(t)/t,t,t=-Pi..Pi],
     coords=polar,title="cochleoid");
```



<u>Ví dụ 2</u> : Vẽ đồ thị hàm số r= 2sin3φ

```
plot([2*sin(3*t),t,t=-2*Pi/3..2*Pi/3]
      ,coords=polar);
```



c) <u>Hàm số ẩn</u>: F(x,y) = 0

Trước khi ra lệnh vẽ phải gọi thủ tục vẽ đồ thị hàm ẩn bởi lệnh:

```
with(plots,implicitplot);
```

và vẽ bằng lệnh :

```
implicitplot(F(x,y),x=a..b,y=c..d);
```

<u>Ví dụ</u>1: Vẽ đồ thị hàm ẩn : $x^3 + y^3 - 3xy = 0$

```
with(plots,implicitplot);
implicitplot(x^3+y^3-3*x*y,x=-2..2,
                  y=-2..2);
```



<u>Ví dụ 2</u>: Vẽ đồ thị $x^2 - y^2 = 1$ và $x^2 + y^2 = 4$

```
with(plots):
k:=implicitplot(x^2 - y^2 = 1,x=-2..2,
    y=-2..2,color=red):
g:=implicitplot(x^2+y^2=4,x=-2..2,
    y=-2..2,color=blue):
display(k,g);
```



d) Hàm nhiều biến:

Dùng lệnh :

plot3d(expr1, x=a..b, y=c..d,options)
plot3d(f, a..b, c..d, options)
plot3d([exprf,exprg,exprh], x=a..b, y=c..d, options)
plot3d([f,g,h], a..b, c..d, options)

trong đó:

expr1, exprf,exprg,exprh là biểu thức chứa x,y
f, g, h là các hàm hai biến.
options bao gồm các lựa chọn sau:

- **coords=c** : chọn hệ tọa độ Descartes, cylindrical(trụ), spherica (cầu)
- **orientation=[theta,phi]**:xoay đồ thị theo các góc theta,phi là cặp tham số (θ,φ) trong tọa độ cầu, giá trị ngầm định [45,45].
- **projection=r** : chọn chiếu phối cảnh với r∈[0,s1], r=0 ('FISHEYE') , r=0.5 ('NORMAL') , giá trị ngầm định (default) là r=1 ('ORTHOGONAL')
- **style=s** : chọn một kiểu vẽ mặt trong các loại sau : POINT, HIDDEN, PATCH (mảnh ghép- default style), WIREFRAME (khung dây), CONTOUR (đường đồng mức), PATCHNOGRID, PATCHCONTOUR, LINE.

Ví dụ 1: Vẽ đồ thị $z = x^2+y^2$ trong miền D : $-10\le x \le 10$ và $-10\le y \le 10$

```
plot3d(x^2+y^2,x=-10..10,y=-10..10,
  orientation=[30,90],axes=normal);
```



```
plot3d(x^2+y^2,x=-10..10,y=- 10..10,
orientation=[30,90],style=wireframe,axes=normal);
```



```
plot3d(x^2+y^2,x=-10..10,y=- 10..10,
orientation=[30,90],style=patchnogrid);
```

Chú ý: * Để xoay đồ thị , ta click chuột vào đồ thị, ấn giữ và di chuyển chuột.
* Để thay đổi các lựa chọn, ta đưa chuột vào đồ thị , ấn nút phải chuột.

e) <u>Mặt dạng tham số </u>: **plot3d([(s,t),g(s,t),h(s,t)],s=a..b,t=c..d);**

Ví dụ :

```
plot3d([r*cos(phi),r*sin(phi),r],r=0..1,
        phi=0..2*Pi);
```



f) **<u>Mặt cong dạng hàm ẩn</u>:**

```
implicitplot3d(z=x*y,x=-2..2,y=-2..2,
                      z=-3..3);
```



g) **<u>Vẽ đa diện (polyhedra):</u>**

Dùng lệnh : **polyhedraplot(L,options);**

Trong đó :

- L là tập hợp, danh sách các điểm $[[x_1,y_1,z_1], [x_2,y_2,z_2], ...]$
- options là **polyscale** = <constant> và **polytype** = <set> trong đó *polyscale* điều khiển kích thước đa diện (ngầm định là 1) , còn *polytype* là kiểu của đa diện nhận giá trị (tetrahedron:tứ diện-giá trị ngầm định, octahedron: bát diện, hexahedron , dodecahedron …)

<u>Ví dụ2</u>: Vẽ tứ diện

```
with(plots):
polyhedraplot([0,0,0],polytype=tetrahedron,
        orientation=[70,60]);
```



<u>Ví dụ2</u>: Vẽ hai tứ diện bằng nhau và bắt đầu từ [0,0,0], [1,1,1]

```
with(plots):
polyhedraplot([[0,0,0],[1,1,1]],
   polytype=tetrahedron,
       polyscale=0.5,orientation=[70,60]);
```



Ví dụ 3: Vẽ hình hộp

```
polyhedraplot([0,0,0],polytype=hexahedron,
 style=PATCH,scaling=CONSTRAINED);
```



h) Vẽ hình trụ (cylinder) : cylinderplot(L,r1,r2,options);

- L là biểu thức có 2 biến r, θ hoặc danh sách có 3 biểu thức [r,θ,z]
- r_1, r_2 là miền của biến có dạng : biến=a..b
- Nếu L là biểu thức có 2 biến r, θ ,z thì r_1 , r_2 là miền của θ và z

```
with(plots):
cylinderplot(1,theta=0..2*Pi,z=-1..1);
```



- Nếu L là danh sách 3 biến [r,θ,z] thì r_1, r_2 là miền của mặt chiếu

```
cylinderplot([z*theta,theta,cos(z^2)],
        theta=0..Pi,z=-2..2);
```



i) Vẽ mặt cầu (sphere) : sphereplot(L,r1,r2,options);

```
with(plots):
sphereplot(1,theta=0..2*Pi,phi=0..Pi);
```



```
sphereplot((5*cos(y)^2 -1)/2,x=0..Pi,
        y=-Pi..Pi,style=PATCH);
```



5. VẼ MẶT CẮT

Để xác định mặt cắt của các mặt cong, ta dùng lệnh **display3d** để vẽ nhiều mặt cong trên cùng hệ trục tọa độ, như ví dụ sau:

<u>Ví dụ:</u> Vẽ hình khối xác định bởi hai hình trụ :

$$x^2 + y^2 = 1 \text{ và } z^2 + y^2 = 1$$

- Đầu tiên ta vẽ hai mặt trụ trong một hệ trục :

```
with(plots):
J:=cylinderplot(1,theta=0..2*Pi,z=-1..1,color=yellow):
K:=plot3d({sqrt(1-y^2),-sqrt(1-y^2)},y=-1..1,
           x=-2..2,color=gold ):
display3d({J,K});
```



- Phần hình trụ ngang chứa trong hình trụ đứng vẽ bằng lệnh:

```
plot3d({sqrt(1-y^2),-sqrt(1-y^2)},
  y=-sqrt(1-x^2)..sqrt(1-x^2),x=-1..1,
  orientation=[45,45],axes=NORMAL);
```



- Phần hình trụ đứng chứa trong hình trụ ngang :

```
plot3d({[x,sqrt(1-x^2),z],[x,-sqrt(1-x^2),z]},
x=-1..1,z=-sqrt(1-x^2)..sqrt(1-x^2),color=gold);
```



- Ghép hai phần giao này ta được vật thể giới hạn bởi hai hình trụ trên.

```
K:= plot3d({sqrt(1-y^2),-sqrt(1-y^2)},
    y=-sqrt(1-x^2)..sqrt(1-x^2),x=-1..1,
    orientation=[45,45],axes=NORMAL,color=gold):
J:=plot3d({[x,sqrt(1-x^2),z],[x,-sqrt(1-x^2),z]},
   x=-1..1,z=-sqrt(1-x^2)..sqrt(1-x^2),
   color=yellow):
display({K,J});
```



<u>Ví dụ</u> : Dựng vật thể giới hạn bởi các mặt :

$$y = \sqrt{x}\ ;\ y = 2\sqrt{x}\ ;\ z = 0\ ;\ x + z = 4$$

```
with(plots):
k:=plot3d([x,sqrt(x),z],x=0..4,z=0..4,
   axes='NORMAL',color=red):
g:=plot3d([x,2*sqrt(x),z],x=0..4,z=0..4,
    orientation=[-47,71],color=green):
h:=plot3d(4-x,x=0..4,y=0..4,style=hidden,
    light=[50,25,1,1,0]):
display({k,g,h});
```



```
k:=plot3d([x,sqrt(x),z],x=0..4,z=0..4-x,axes='NORMAL',
     color=red):
g:=plot3d([x,2*sqrt(x),z],x=0..4,z=0..4-x,color=green,
    orientation=[1,60]):
h:=plot3d(4-x,x=0..4,y=sqrt(x)..2*sqrt(x),color=yellow):
n:=plot3d(0,x=0..4,y=sqrt(x)..2*sqrt(x),color=white):
display({k,g,h,n});
```



Thực hành

1. Cho hàm số f(x)= $\begin{cases} x\sin\left(\dfrac{1}{x}\right) & \text{khi } x \neq 0 \\ 1 & \text{khi } x = 0 \end{cases}$
   a. Tính giá trị f(1), f(2) , f(3)
   b. Vẽ đồ thị hàm số f(x)
2. Vẽ đồ thị hai hàm số sau đây trên cùng hệ trục tọa độ và tìm tọa độ giao điểm : $f(x) = \dfrac{x}{1+x^2}$; $g(x) = x^2 - x - 2$
3. Vẽ đồ thị các hàm so:
   a. $z = \dfrac{x}{x^2 + y^2}$
   b. $z = x(x^2 - 3y^2)$

4. Vẽ đồ thị các hàm số:
   a. $x = \dfrac{\cos t}{1+\sin^2 t}$, $y = \dfrac{\cos t.\sin t}{1+\sin^2 t}$, $-\pi \le t \le \pi$
   b. $r^2 = \cos 2\varphi$
   c. $(x^2+y^2)^2 = x^2 - y^2$
   d. $t \mapsto \left(\dfrac{t^2-1}{t^2+1}, \dfrac{2t}{t^2+1}\right)$, $t\in(-\infty , +\infty)$
5. Hãy viết thủ thục trong Maple để tính đa thức Legrendre $L_n(x)$ được định nghĩa bởi : $L_0(x)=1$, $L_1(x) = x$
   và $L_n(x) = \dfrac{n-1}{n}\left(x.L_{n-1}(x) - L_{n-2}(x)\right) + L_{n-1}(x)$ khi n>1
   a) Tính $L_7(x)$
   b) Vẽ đồ thị $L_2(x)$, $L_3(x)$ trên cùng một hệ trục tọa độ.
6. Cho $f(x) = \dfrac{x}{x^2+1}$. Tính $f^2(x)$, $f^3(x)$ và vẽ đồ thị của chúng.

   Vẽ vật thể giới hạn bởi các mặt:
7. $y= x^2$, $z = y$, $z +y =2$
8. $x^2 + y^2 = 4$, $z = -2$, $y + z = 2$
9. $x^2 + y^2 + z^2 = 4$, $x^2 + y^2 = 1$
10. $x^2 + y^2 - 4z^2 + 4 =0$, $z = 2$

GIỚI HẠN – LIÊN TỤC

I. GIỚI HẠN

1) Giới hạn hàm một biến

Câu lệnh : **limit(f, x=a); limit(f, x=a, dir);**

f - một biểu thức đại số (an algebraic expression)
x - một tên (a name)
a - một biểu thức đại số (điểm giới hạn , có thể infinity, -infinity)
dir – (tùy chọn) hướng lấy giới hạn là : left, right, real, complex

Ví dụ 1: Tính $\lim_{x\to 0} \cos(x)^{\frac{1}{x}}$

```
limit(cos(x)^(1/x),x=0);
```
$\rightarrow$ **1**

Vì Maple phân biệt chữ hoa và chữ thường, nên lệnh :

Limit(f(x) , x=a) cho kí hiệu : $\lim_{x\to a} f(x)$

Do đó lệnh :

```
Limit(cos(x)^(1/x),x=0)=
    limit(cos(x)^(1/x),x=0);
```

$$\lim_{x\to 0} \cos(x)^{\left(\frac{1}{x}\right)} = 1$$

Trong trường hợp giới hạn hai phía không xác định, ta xét giới hạn bên phải, bên trái :

Ví dụ 2: Tính $\lim_{x\to 0} e^{\frac{1}{x}}$

```
Limit(exp(1/x),x=0): %= value(%);
```

$$\lim_{x\to 0} \mathrm{e}^{\left(\frac{1}{x}\right)} = \mathit{undefined}$$

```
Limit(exp(1/x),x=0,right): %= value(%);
```

$$\lim_{x\to 0+} \mathrm{e}^{\left(\frac{1}{x}\right)} = \infty$$

```
Limit(exp(1/x),x=0,left): %= value(%);
```

$$\lim_{x\to 0-} \mathrm{e}^{\left(\frac{1}{x}\right)} = 0$$

Trong trường hợp có tham số, ta cần xác định tham số mới tính được giới hạn :

```
limit(exp(a*x)*cos(b*x),x=-infinity);
```

$$\lim_{x\to(-\infty)} \mathrm{e}^{(a\,x)} \cos(b\,x)$$

```
assume(a>0):
limit(exp(a*x)*cos(b*x),x=-infinity);
```
$\rightarrow$ 0

2) Giới hạn hàm nhiều biến

Câu lệnh : **limit(f, points)**

limit(f, points, dir)

trong đó : f – một biểu thức đại số chứa x , y …

points – tập hợp các đẳng thức dạng { x=a , y=b … }

dir – (tùy chọn) hướng lấy giới hạn

Ví dụ 1: Tính $\lim\limits_{\substack{x\to 0\\ y\to 0}} \dfrac{x^2 - y^2}{x^2 + y^2}$

```
limit((x^2-y^2)/(x^2+y^2), {x=0,y=0});
```

undefined

```
limit((x^2-y^2)/(x^2+y^2), {x=t,y=2*t});
```

$$\frac{-3}{5}$$

Ví dụ 2: $\lim\limits_{\substack{x\to 0\\ y\to 0}} \dfrac{xy}{3-\sqrt{xy+9}}$

```
limit((x*y)/(3-sqrt(x*y+9)), {x=0,y=0});
```

$$\operatorname{limit}\left(\frac{x\,y}{3-\sqrt{x\,y+9}}, \{x=0, y=0\}\right)$$

Đổi biến t=xy ta được :

```
limit((t)/(3-sqrt(t+9)), {t=0});   -> -6
```

II. HÀM SỐ LIÊN TỤC

1) Kiểm tra hàm liên tục trên khoảng số thực (a,b)

iscont(expr, x = a .. b)
iscont(expr, x = a .. b, 'closed')
iscont(expr, x = a .. b, 'open')

Trong đó :

expr - một biểu thức đại số chứa biến x
x - tên biến
a..b - khoảng số thực với a,b là hằng số hay infinity, -infinity
'closed' – tùy chọn kiểm tra trong đoạn [a,b]
'open' - tùy chọn ngầm định (default) là khoảng (a,b)

Ví dụ :

```
iscont( 1/x, x=-1..1 );
```

false

```
iscont( 1/x, x=0..1 );
```

true

```
iscont( 1/x, x=0..1, 'closed' );
```

false

2) Điểm gián đoạn

Để tìm điểm gián đoạn thực ta dùng lệnh : **discont(f, x)**
trong đó : f – một biểu thức đại số chứa biến x

- discont cho ra tập hợp các giá trị thực mà tại đó có thể hàm gián đoạn, kể cả $\pm\infty$.
- Khi kết quả cho bởi :
  - _Zn~ : mọi số nguyên thuộc Z, còn n số thứ tự.
  - _NNn~ : mọi số nguyên không âm thuộc N
  - _Bn~ : tập hợp nhị phân {0,1}

Ví dụ : Tìm điểm gián đoạn của $f(x)=\dfrac{x}{x^2-x-6}$

```
discont(x/(x^2-x-6),x);
```

$\{-2, 3\}$

Ví dụ : Tìm điểm gián đoạn của $f(x)=\dfrac{1}{\sin x}$

```
discont(1/sin(x),x);
```

$\{\pi\ \_Z1\sim\}$ nghĩa là $k\pi$, $k\in Z$

Ví dụ : Tìm điểm gián đoạn của $f(x)=\dfrac{1}{\sin x-\frac{1}{2}}$

```
discont(1/(sin(x)-1/2),x);
```

$\{\frac{1}{6}\pi+\frac{2}{3}\pi\ \_B1\sim+2\pi\ \_Z3\sim\}$

nghĩa là : $\frac{1}{6}\pi+\frac{2}{3}\pi.\{0,1\}+2\pi k$, $k\in Z$

Ví dụ : Cho hàm số f(x) = $\begin{cases} x^2 & \text{khi } x\le 3 \\ ax+1 & \text{khi } x>3 \end{cases}$

a) xác định a để hàm số liên tục trên R.
b) Vẽ đồ thị hàm số f(x) với giá trị của a vừa tìm được.

Giải:

```
# a)
restart: f:=x->piecewise(x<=3,x^2,a*x+3);
```

$f := x \to \text{piecewise}(x\le 3, x^2, a\,x+3)$

```
pt:=limit(f(x),x=3,left)-limit(f(x),x=3,right);
```

$pt := 6-3\,a$

```
a=solve(pt,a);
```

$a=2$

```
# b)
g:=x->piecewise(x<=3,x^2,2*x+3);
```

$g := x \to \text{piecewise}(x\le 3, x^2, 2\,x+3)$

```
plot(g(x),x=-6..6);
```



Thực hành

1) Tính các giới hạn sau:

a) $\lim\limits_{x\to 0}\dfrac{\sin 3x}{x}$ b) $\lim\limits_{x\to 0}(\sin x)^{\frac{1}{x}}$ c) $\lim\limits_{x\to 0}\left(\dfrac{1-\cos 2x}{3x^2}\right)$

d) $\lim\limits_{x\to\infty}\left(1+\dfrac{\pi}{x}\right)^x$ e) $\lim\limits_{x\to 0} x^{\sin x}$ f) $\lim\limits_{x\to\infty}(2^x+3^x)^{\frac{1}{x}}$

2) Tính các giới hạn sau:

a) $\lim\limits_{x\to\infty}\dfrac{\ln x}{x}$ b) $\lim\limits_{x\to\infty}\dfrac{\ln x}{e^x}$ c) $\lim\limits_{x\to\infty}\dfrac{x^2+\sin x}{2x^2+\cos 4x}$

d) $\lim\limits_{x\to 0}\dfrac{2}{1+e^{-\frac{1}{x}}}$ e) $\lim\limits_{x\to\infty} sh(th(x)) - th(sh(x))$

3. Tìm tham số A để hàm liên tục tại x = 1

a) $f(x)=\begin{cases}(1-x)\operatorname{tg}\left(\dfrac{\pi x}{2}\right) & \text{khi } x\neq 1\\ A & \text{khi } x=1\end{cases}$

b) $g(x)=\begin{cases}x+1 & \text{khi } x\le 1\\ 3-Ax^2 & \text{khi } x>1\end{cases}$

4. Chứng minh rằng hàm số $y=\dfrac{1}{1+2^{\frac{1}{x}}}$ gián đoạn loại 1 tại x = 0.

Vẽ đồ thị hàm số y tại lân cận x = 0

5. Tìm điểm gián đoạn của hàm số:

a) $f(x)=\dfrac{\operatorname{tg}(\pi x)}{x-2}$ b) $g(x)=\dfrac{1}{1-e^{\frac{1}{x-1}}}$ c) $h(x)=\dfrac{1-\sqrt{\cos x}}{\sin^2 x}$

6. Chứng tỏ phương trình: $x.2^x=1$

có nghiệm bằng cách dựng đồ thị.

Giải gần đúng nghiệm của phương trình trên.

ĐẠO HÀM

I. ĐẠO HÀM

1. Đạo hàm cấp 1

Câu lệnh :

- **diff(f(x) ,x);** tính đạo hàm cấp 1 của hàm số f(x)
- **Diff(f(x),x);** In ký hiệu đạo hàm f'(x) là $\frac{\partial}{\partial x} f(x)$

Ví dụ1 : Tính đạo hàm của y= $\ln\left(\frac{x}{x^2+1}\right)$

```
Diff(log(x/(x^2+1)),x)=
    normal(diff(log(x/(x^2+1)),x));
```

$$\frac{\partial}{\partial x}\ln\left(\frac{x}{x^2+1}\right) = -\frac{x^2-1}{x\,(x^2+1)}$$

Ví dụ2: Tính đạo hàm của y = $\sqrt{x\sqrt{x}}$

```
diff(sqrt(x*sqrt(x)),x);
```

$$\frac{3}{4}\frac{\sqrt{x}}{\sqrt{x^{(3/2)}}}$$

Ví dụ 3: Tính đạo hàm của f(x) = $\begin{cases} x^2+1 & \text{khi } x<1 \\ 3-x & \text{khi } x\ge 1 \end{cases}$

```
f:=x->piecewise(x<1,x^2+1,3-x);
```

$$f := x \to \text{piecewise}(x<1,\ x^2+1,\ 3-x)$$

```
diff(f(x),x);
```

$$\begin{cases} 2x & x<1 \\ \textit{undefined} & x=1 \\ -1 & 1<x \end{cases}$$

2. Đạo hàm cấp n

- Đạo hàm cấp 2 : **diff(f(x) , x, x);** hay **diff(f(x) ,x$2)**
- Đạo hàm cấp n : **diff(f(x) , x$n) ;**

với n số nguyên dương *xác định*.

Ví dụ 1: Tính đạo hàm cấp 2 của y = $\frac{x}{x+1}$

```
diff(x/(x+1),x,x);
```

$$-2\frac{1}{(x+1)^2}+\frac{2x}{(x+1)^3}$$

Rút gọn bởi lệnh : `normal(%);`

$$-2\frac{1}{(x+1)^3}$$

Ví dụ 2 : Tính đạo hàm cấp 2 của f(x)= $\begin{cases} x^2+1 & \text{khi } x<1 \\ 3-x & \text{khi } x\ge 1 \end{cases}$

```
f:=x->piecewise(x<1,x^2+1,3-x):
diff(f(x),x$2);
```

$$\begin{cases} 2 & x<1 \\ \textit{undefined} & x=1 \\ 0 & 1<x \end{cases}$$

Ví dụ 3: Tìm công thức qui nạp cho đạo hàm cấp n của hàm số

$y=\frac{1}{x^2-1}$. Suy ra $y^{(n)}$

- Định nghĩa hàm số đạo hàm

```
f:=n->diff((x^2-1)*y(x)=1,x$n);
```

$$f := n \to \text{diff}((x^2-1)\,\text{y}(x)=1,\ x\,\$\,n)$$

- Hiện dãy đạo hàm, tìm công thức qui nạp

```
f(2);f(3); f(4);
```

$$2\,\text{y}(x)+4\,x\left(\frac{\partial}{\partial x}\text{y}(x)\right)+(x^2-1)\left(\frac{\partial^2}{\partial x^2}\text{y}(x)\right)=0$$

$$6\left(\frac{\partial}{\partial x}\text{y}(x)\right)+6\,x\left(\frac{\partial^2}{\partial x^2}\text{y}(x)\right)+(x^2-1)\left(\frac{\partial^3}{\partial x^3}\text{y}(x)\right)=0$$

$$12\left(\frac{\partial^2}{\partial x^2}\text{y}(x)\right)+8\,x\left(\frac{\partial^3}{\partial x^3}\text{y}(x)\right)+(x^2-1)\left(\frac{\partial^4}{\partial x^4}\text{y}(x)\right)=0$$

- Công thức qui nạp :

```
n*(n-1)*y(n-2)+2*n*x*y(n-1)+(x^2-1)*y(n)
```

- Giải phương trình đệ qui ta được đạo hàm cấp n :

```
rsolve({n*(n-1)*y(n-2)
       +2*n*x*y(n-1)+(x^2-1)*y(n),
        y(0)=1/(x^2-1),
        y(1)=-2*x/(x^2-1)^2},y);
```

$$\frac{1}{2}\,\frac{\left(-\frac{1}{x-1}\right)^n\Gamma(n+1)}{x-1}-\frac{1}{2}\,\frac{\left(-\frac{1}{x+1}\right)^n\Gamma(n+1)}{x+1}$$

trong đó $\Gamma(n+1)=n!$

Ví dụ 4 : Tính đạo hàm cấp n của hàm số $y = xe^x$

- Định nghĩa hàm số đạo hàm cấp n:

```
dh:=n->diff(x*exp(x),x$n);
```

$$dh := n \to \text{diff}(x\,\text{e}^x,\ x\,\$\,n)$$

- Liệt kê các đạo hàm cấp 1, cấp 2 :

```
factor(dh(1));factor(dh(2));
```

$$\text{e}^x\,(1+x) \qquad \text{e}^x\,(2+x)$$

- Giả sử $y^{(n)} = e^x(n + x)$, ta đạo hàm hai vế được :

```
y^(n+1)=factor(diff((n+x)*exp(x),x));
```

$$y^{(n+1)}=\text{e}^x\,(1+n+x)$$

Vậy $y^{(n)} = e^x(n + x)$

3. Đạo hàm riêng

Câu lệnh : **diff(f(x,y) , xn, ym);**

Ví dụ : Tính đạo hàm riêng cấp ba : $\frac{\partial^3}{\partial x \partial y^2}\left(\frac{x+y}{x^2+y^2}\right)$

```
diff((x+y)/(x^2+y^2),x,y$2);
```

$$8\frac{y^2}{(x^2+y^2)^3}-\frac{2}{(x^2+y^2)^2}+\frac{16\,x\,y}{(x^2+y^2)^3}-\frac{48\,(x+y)\,x\,y^2}{(x^2+y^2)^4}+\frac{8\,(x+y)\,x}{(x^2+y^2)^3}$$

```
normal(%);
```

$$6\,\frac{-6\,y^2\,x^2+y^4+x^4+4\,x^3\,y-4\,x\,y^3}{(x^2+y^2)^4}$$

4. Toán tử đạo hàm

- D(f) : Hàm số đạo hàm của hàm một biến f .
  Nghĩa là D(f)(x) = diff(f(x), x)
- D[i](f) : Hàm số đạo hàm riêng theo biến thứ i.

Tương tự : D[i, j](f) = D[i](D[j](f))

Ví dụ 1: Cho f(x) = x.arcsin(x) . Tính f'(x) , f'(1/2).

```
f:=x->x*arcsin(x);
```

$$f := x \to x\,\arcsin(x)$$

```
D(f)(x);
```

$$\arcsin(x)+\frac{x}{\sqrt{1-x^2}}$$

```
D(f)(1/2);
```

$$\frac{1}{6}\pi+\frac{1}{6}\sqrt{3}\,\sqrt{4}$$

Ví dụ 2: Tính đạo hàm riêng cấp 2 của z = xsiny

```
g:=(x,y)-> x*sin(y);
Diff([g(x,y)],x,x)=D[1,1](g)(x,y);
```

$$\frac{\partial^2}{\partial x^2}[x\sin(y)]=0$$

```
Diff([g(x,y)],x,y)=D[1,2](g)(x,y);
```

$$\frac{\partial^2}{\partial y\,\partial x}[x\sin(y)]=\cos(y)$$

```
Diff([g(x,y)],y,x)=D[2,1](g)(x,y);
```

$$\frac{\partial^2}{\partial x\,\partial y}[x\sin(y)]=\cos(y)$$

```
Diff([g(x,y)],y,y)=D[2,2](g)(x,y);
```

$$\frac{\partial^2}{\partial y^2}[x\sin(y)]=-x\sin(y)$$

Ví dụ 3: Cho h(x,y,z) = $\frac{1}{\sqrt{x^2+y^2+z^2}}$

Chứng minh rằng: $\frac{\partial^2 h}{\partial x^2}+\frac{\partial^2 h}{\partial y^2}+\frac{\partial^2 h}{\partial z^2}=0$

```
h:=(x,y,z)->1/(x^2+y^2+z^2)^(1/2);
```

$$h := (x, y, z) \to \frac{1}{\sqrt{x^2+y^2+z^2}}$$

```
normal((D[1,1]+D[2,2]+D[3,3])(h)(x,y,z));
```

$$0$$

5. Đạo hàm hàm số ẩn

```
implicitdiff(f, y, x)
implicitdiff(f, y, x1,...,xk)
```

trong đó :

f	- biểu thức đại số hoặc những phương trình hàm ẩn.
y	- tên biến hoặc tên hàm của biến độc lập.
x, x1, ..., xk	- tên của biến đạo hàm.

- Lệnh **implicitdiff(f,y,x)** tính dy/dx, đạo hàm riêng của hàm y đối với x. Tham số f phải là phương trình của x, y hoặc một biểu thức đại số bằng 0.

Ví dụ 1: Tính đạo hàm của hàm ẩn y đối với x thỏa : $x^2y + y^2 = 1$

```
dy/dx=implicitdiff(x^2*y+y^2=1,y,x);
```

$$\frac{dy}{dx} = -2\frac{x\,y}{x^2 + 2\,y}$$

Giải bằng cách khác :

```
alias(y=y(x)):
diff(x^2*y+y^2=1,x);
```

$$2\,x\,y + x^2\left(\frac{\partial}{\partial x}y\right) + 2\,y\left(\frac{\partial}{\partial x}y\right) = 0$$

```
solve(%,diff(y,x));
```

$$-2\frac{x\,y}{x^2 + 2\,y}$$

- Lệnh **implicitdiff(f, y, x1,...,xk)** tính đạo hàm riêng của y theo các biến x1, x2, ...

Ví dụ 2: Cho phương trình $x^2 + y^2 + z^2 = 1$. Tính $\frac{\partial z}{\partial x}$, $\frac{\partial^2 z}{\partial x^2}$

```
implicitdiff(x^2 + y^2 +z^2=1,z,x);
```

$$-\frac{x}{z}$$

```
implitdiff(x^2 + y^2 +z^2=1,z,x,x);
```

$$-\frac{z^2 + x^2}{z^3}$$

- Lệnh **implicitdiff(F, Y, U, X)** trong đó :
  - F : danh sách phương trình ẩn dạng { f1, f2, … fk }
  - Y : danh sách hàm số dạng {y1, y2 ,…,yn}
  - U : danh sách các hàm cần tính đạo hàm {u1,u2…}
  - X: danh sách biến tính đạo hàm.

Ví dụ 3: Cho phương trình $x^2 + y = z$ và $x+y+z = 1$.

Tính $\frac{\partial z}{\partial x}$, $\frac{\partial y}{\partial x}$

```
implicitdiff({x^2+y=z,x+y+z=1},{y,z},{y,z},x);
```

$$\{\,\mathrm{D}(z) = x - \frac{1}{2},\ \mathrm{D}(y) = -\frac{1}{2} - x\,\}$$

Ví dụ 4: Cho hàm số dạng tham số : x= asint , y=bcos^{2}t .

Tính y'_x

```
pt:={x=a*sin(t),y=b*cos(t)^2}:
implicitdiff(pt,{y,t},y,x)
```

$$-2\frac{b\sin(t)}{a}$$

II. CỰC TRỊ

Lệnh `extrema(expr,constraints,vars,s);`

cho kết quả giá trị cực trị của `expr`

- `expr` : biểu thức tính cực trị.
- `constraints` : tập hợp các phương trình ràng buộc.
- `vars` : tập hợp các biến.
- `s` : danh sách chi tiết các điểm cực trị.

1) Cực trị tự do : có điều kiện ràng buộc `constraints` là `{}`

Lệnh : `extrema(expr,{},vars,s);`

Ví dụ 1 : Tính cực trị của hàm số y = 3x - x^3

```
extrema(3*x-x^3,{},x,s);  ->
```
$\{-2, 2\}$

```
s;  ->
```
$\{\{x=-1\}, \{x=1\}\}$

Hàm số y đạt cực trị tại x= -1 là -2 và tại x= 1 là 2

Ví dụ 2: Tìm cực trị hàm số z = x^2 + y^2 –xy –x -y

```
restart:
z:=(x,y)->x^2+y^2-x*y-x-y:
extrema(z(x,y),{},{x,y},s); ->
```
$\{-1\}$

```
s;    ->
```
$\{\{y=1, x=1\}\}$

Hàm số z đạt cực trị -1 tại (1,1) .

2) Cực trị ràng buộc

Lệnh : `extrema(expr,constraints,vars,s);`

Ví dụ 3: Tính cực trị hàm z = 3x + 4y với điều kiện x^2 + y^2 =1

```
restart:
extrema(3*x+4*y,x^2+y^2-1,{x,y},s);
```
$\{-5, 5\}$

```
s;
```
$\{\{x=\frac{3}{5}, y=\frac{4}{5}\}, \{x=\frac{-3}{5}, y=\frac{-4}{5}\}\}$

Ví dụ 4: Tính cực trị hàm u = x^2 + y^2 + z^2 với điều kiện:

x^2 + y^2 =1 và x+y+z =1

```
restart:
dk:={x^2+y^2=1,x+y+z=1}:
extrema(x^2+y^2+z^2,dk,{x,y,z},s);
s;
```

3) Max , Min

Lệnh: `maximize(expr,option1,location);`

`minimize(expr,option1,location);`

- `expr` : biểu thức cần tìm Min, Max
- `option1` : dạng a..b ; c..d
- `location`: cho điểm đạt Min, Max

Ví dụ 1: Tìm giá trị lớn nhất , nhỏ nhất của hàm số :

y= x^4 – x^2 trên đoạn [-3 , 3]

```
minimize(x^4 - x^2, x=-3..3, location);
```
$$\frac{-1}{4}, \left\{\left[\{x=-\frac{1}{2}\sqrt{2}\}, \frac{-1}{4}\right], \left[\{x=\frac{1}{2}\sqrt{2}\}, \frac{-1}{4}\right]\right\}$$

```
maximize(x^4 - x^2, x=-3..3, location);
```
$$72, \{[\{x=-3\}, 72], [\{x=3\}, 72]\}$$

Ví dụ 2: Tìm giá trị lớn nhất , nhỏ nhất của hàm số :

$z = x^2 - 3x + y^2 + 3y + 3$ trong miền $2 \le x \le 4$ và $-4 \le y \le -2$

```
minimize(x^2-3*x+y^2+3*y+3, x=2..4,
        y=-4..-2, location);
```

$-1, \{[\{y=-2, x=2\}, -1]\}$

```
maximize(x^2-3*x+y^2+3*y+3, x=2..4,
         y=-4..-2, location);
```

$11, \{[\{x=4, y=-4\}, 11]\}$

Ví dụ 3: Tìm giá trị lớn nhất của hàm số $u = -x + y + 2z$

thỏa các điều kiện :

$$\begin{cases} 3x + 4y - 3z \le 23 \\ 5x - 4y - 3z \le 10 \\ 7x + 4y + 11z \le 30 \\ x \ge 0; y \ge 0; z \ge 0 \end{cases}$$

```
restart:with(simplex):
u:= -x + y + 2*z:
dk:= {3*x+4*y-3*z <= 23, 5*x-4*y-3*z <= 10,
                         7*x+4*y+11*z <= 30}:
maximize(u,dk,NONNEGATIVE); ->
```

$\{x=0, y=\frac{49}{8}, z=\frac{1}{2}\}$

```
Max=subs(%,u);    ->
```

$Max = \frac{57}{8}$

Thực hành

1) Tính đạo hàm :

a) $y = \frac{1}{x} + \frac{1}{\sqrt{x}} + \frac{1}{\sqrt[3]{x}}$ b) $y = \begin{cases} x & \text{khi } x < 0 \\ \ln(1+x) & \text{khi } x \ge 0 \end{cases}$

2) Tính đạo hàm cấp cao tương ứng:

a) $y = e^{\sin x}.\cos(\sin x)$. Tính y'' b) $y = \frac{x^2}{1-x}$. Tính $y^{(8)}$

3) Tính đạo hàm cấp n của các hàm số :

a) $y = \frac{1}{x^2 + 4x + 3}$

b) $y = x^2 e^x$

4) Tính đạo hàm $\frac{dy}{dx}$ biết :

a) $\begin{cases} x = a(t - \sin t) \\ y = b(1 - \cos t) \end{cases}$ b) $\operatorname{arctg}\frac{y}{x} = \ln\sqrt{x^2 + y^2}$

5) Cho hàm số thực g xác định bởi :

$$g(x,y,z) = \frac{1}{\sqrt{(x-a)^2 + (y-b)^2 + (z-c)^2}}$$

Chứng tỏ g là nghiệm của phương trình :

$$\frac{\partial^2 g}{\partial x^2} + \frac{\partial^2 g}{\partial y^2} + \frac{\partial^2 g}{\partial z^2} = 0$$

6) Tính y', y'' của hàm số ẩn định nghĩa bởi $\sqrt{x} + \sqrt{y} = 1$.

7) Tìm cực trị của hàm số :

a) $y = \frac{3x^2 + 4x + 1}{x^2 + x + 1}$ b) $z = x^3 + 3xy^2 - 15x - 12y$

c) u = xyz thỏa điều kiện : x + y + z = 5 và xy + yz + zx = 8

8) Tìm giá trị lớn nhất và nhỏ nhất của hàm số :

a) $y = \sqrt[3]{(x^2 - 2x)^2}$ trên [0,3]

b) $z = x^2 + y^2 - xy + x + y$ trong miền $x \le 0$; $y \le 0$ và $x + y + 3 \ge 0$

9) Tìm khoảng cách ngắn nhất từ mặt $x^2 + y^2 - z^2 = 1$ đến gốc O.

10) Tìm trên ellip $x^2 + 9y^2 = 9$ các điểm gần nhất và xa nhất đối với đường thẳng $4x + 9y = 16$.

TÍCH PHÂN

I. TÍCH PHÂN

Câu lệnh :

int(expr, x) : Tích phân bất định
int(expr, x=a..b, ...) : Tích phân xác định
Int(expr, x) : Ký hiệu tích phân bất định
Int(expr, x=a..b, ...) : Ký hiệu tích phân xác định

Trong đó :

expr – một biểu thức đại số
x – tên biến tích phân
a, b – cận tích phân
... – tùy chọn.

Ví dụ 1: Tính tích phân bất định $\int \frac{x}{x^3+1}dx$

```
Int( x/(x^3+1), x );
```

$$\int \frac{x}{x^3+1}\,dx$$

```
int( x/(x^3+1), x );
```

$$-\frac{1}{3}\ln(x+1)+\frac{1}{6}\ln(x^2-x+1)+\frac{1}{3}\sqrt{3}\arctan\left(\frac{1}{3}(2x-1)\sqrt{3}\right)$$

```
Int( x/(x^3+1), x ): %=value(%);
```

$$\int \frac{x}{x^3+1}\,dx=-\frac{1}{3}\ln(x+1)+\frac{1}{6}\ln(x^2-x+1)+\frac{1}{3}\sqrt{3}\arctan\left(\frac{1}{3}(2x-1)\sqrt{3}\right)$$

Ví dụ 2: Tính tích phân xác định $\int_0^1 \frac{x}{\sqrt{x}+1}dx$

```
Int(x/(sqrt(x)+1),x=0..1): % =value(%);
```

$$\int_0^1 \frac{x}{\sqrt{x}+1}\,dx=\frac{5}{3}-2\ln(2)$$

Ví dụ 3: Tính tích phân suy rộng $\int_1^{+\infty} \frac{1}{x^2+2x+1}dx$

```
Int(1/(x^2+2*x+1),x=1..+infinity):
% = value(%);
```

$$\int_1^{\infty} \frac{1}{x^2+2x+1}\,dx=\frac{1}{2}$$

□ Trong trường hợp tích phân không biểu diễn được dưới dạng hàm sơ cấp thì kết quả tích phân được biểu diễn qua các hàm :

```
erf(x) = 2/sqrt(Pi) * int(exp(-t^2), t=0..x)
FresnelC(x) =  int(cos(Pi/2*t^2), t=0..x);
FresnelS(x) =  int(sin(Pi/2*t^2), t=0..x);
EllipticF(z,k)=int(1/sqrt(1-t^2)/sqrt(1-k^2*t^2),t=0..z)
EllipticE(z,k) = int(sqrt(1-k^2*t^2)/sqrt(1-t^2),t=0..z)
EllipticPi(z,nu,k)=int(1/(1-nu*t^2)/sqrt(1-t^2)/
                                  sqrt(1-k^2*t^2),t=0..z)
Si(x)   = int(sin(t)/t, t=0..x)
Ci(x)   = gamma + ln(x) + int((cos(t)-1)/t, t=0..x)
Ssi(x)  = Si(x) - Pi/2
Shi(x)  = int(sinh(t)/t, t=0..x)
Chi(x)  = gamma + ln(x) + int((cosh(t)-1)/t, t=0..x)
```

*Chú ý : Dùng lệnh **evalf** để tính gần đúng các giá trị tích phân xác định được biểu diễn qua các hàm đặc biệt trên.*

Ví dụ 4: Tính $\int e^{-x^2} dx$

```
int(exp(-x^2),x);
```

$$\frac{1}{2}\sqrt{\pi}\,\mathrm{erf}(x)$$

Ví dụ 5: Tính tích phân suy rộng : $\int_{-1}^{1} \frac{1}{x^2} dx$

```
Int(1/x^2,x=-1..1):
%=value(%);
```

$$\int_{-1}^{1} \frac{1}{x^2}\, dx = \infty$$

Ví dụ 6: Tính $\int \sin(x^2) dx$

```
Int(sin(x^2),x):
%=value(%);
```

$$\int \sin(x^2)\, dx = \frac{1}{2}\sqrt{2}\,\sqrt{\pi}\,\mathrm{FresnelS}\left(\frac{\sqrt{2}\,x}{\sqrt{\pi}}\right)$$

Ví dụ 7: Tính gần đúng $\int_{0}^{1} \cos(x^2) dx$

```
int(cos(x^2),x=0..1);
```

$$\frac{1}{2}\mathrm{FresnelC}\left(\frac{\sqrt{2}}{\sqrt{\pi}}\right)\sqrt{2}\,\sqrt{\pi}$$

```
evalf(%);
```

.9045242380

II. TÍCH PHÂN BỘI

1. *TÍCH PHÂN KÉP*

Để tính tích phân kép, trước tiên ta dùng lệnh : **with(student)** để mở tiện ích **student** cho phép dùng các lệnh sau :

- **Doubleint(g, x, y);**
- **Doubleint(g, x=a..b, y=c..d);**
- **Doubleint(g, x, y, Domain);**

Trong đó :

- g : biểu thức cần tích phân
- x, y : biến tích phân và *chú ý đến thứ tự biến lấy tích phân*.
- Domain : tên miền tích phân

Để tính giá trị của tích phân kép ta dùng lệnh **%=value(%);** sau lệnh tích phân kép.

Ví dụ 1: Tính $\int_0^1 \int_x^1 x + y \, dy \, dx$

```
with(student):
Doubleint(x+y,y=x..1,x=0..1):
%=value(%);
```

$$\int_0^1 \int_x^1 x + y \, dy \, dx = \frac{1}{2}$$

Ví dụ 2: Tính $\int_x^1 \int_0^1 x + y \, dx \, dy$

```
Doubleint(x+y,x=0..1,y=x..1):
%=value(%);
```

$$\int_x^1 \int_0^1 x + y \, dx \, dy = 1 - \frac{1}{2} x - \frac{1}{2} x^2$$

2. *TÍCH PHÂN BỘI BA*

Tương tự tích phân kép chỉ khác dùng lệnh :

Tripleint(g, x, y, z)
Tripleint(g, x = a..b, z = e..f, y = c..d)
Tripleint(g, x, y, z, Domain)

Ví dụ 1:

Tính $\iiint_V x^3 y^2 z dxdydz$ với V : 0≤ x ≤1 , 0≤ y ≤ x , 0 ≤ z ≤ xy

```
with(student):
Tripleint(x^3*y^2*z,z=0..x*y,y=0..x,x=0..1);
```

$$\int_0^1 \int_0^x \int_0^{xy} x^3 y^2 z \, dz \, dy \, dx$$

```
%=value(%);
```

$$\int_0^1 \int_0^x \int_0^{xy} x^3 y^2 z \, dz \, dy \, dx = \frac{1}{110}$$

Ví dụ 2: Tính $\iiint_V \sqrt{x^2+y^2+z^2}\,dxdydz$ với V là hình cầu tâm 0 , bán kính R

Ta dùng tọa độ cầu :

```
Tripleint(r^3*cos(theta),r=0..R,
          theta=-Pi/2..Pi/2,phi=0..2*Pi);
```

$$\int_0^{2\pi}\int_{-1/2\,\pi}^{1/2\,\pi}\int_0^R r^3\cos(\theta)\,dr\,d\theta\,d\phi$$

```
%=value(%);
```

$$\int_0^{2\pi}\int_{-1/2\,\pi}^{1/2\,\pi}\int_0^R r^3\cos(\theta)\,dr\,d\theta\,d\phi = R^4\,\pi$$

3. *TÍCH PHÂN ĐƯỜNG*

Lineint(f(x,y), x=x(t), y=y(t))
Lineint(f(x,y), x, y =a..b)
Lineint(f(x,y,z), x, y, z)

Trong đó :

f(x,y) – một biểu thức chứa x và y và biến .
a, b – (tuỳ chọn) cận dưới và cận trên

Ví dụ 1: Tính tích phân đường loại 1: $\int_{AB} z^2 dl$ với AB xác định bởi: x= acost , y= asint, z= bt và 0 $\le$ t $\le$ 3

```
with(student):
Lineint(z^2,x=a*cos(t),y=a*sin(t),z=b*t,t=0..3);
```

$$\int_0^3 b^2 t^2 \sqrt{\left(\frac{\partial}{\partial t} a\cos(t)\right)^2 + \left(\frac{\partial}{\partial t} a\sin(t)\right)^2 + \left(\frac{\partial}{\partial t} b\,t\right)^2}\;dt$$

```
%=value(%);
```

$$\int_0^3 b^2 t^2 \sqrt{a^2\sin(t)^2 + a^2\cos(t)^2 + b^2}\;dt = 9\,b^2\sqrt{a^2+b^2}$$

Ví dụ 2: Tính tích phân đường : $\int_{AB} xydl$ với AB là cung ellip $x^2+\frac{y^2}{4}=1, x\ge 0, y\ge 0$

```
with(student):
Lineint(x*y,x=cos(t),y=2*sin(t),t=0..Pi/2);
```

$$\int_0^{1/2\,\pi} 2\cos(t)\sin(t)\sqrt{\left(\frac{\partial}{\partial t}(2\sin(t))\right)^2 + \left(\frac{\partial}{\partial t}\cos(t)\right)^2}\;dt$$

```
%=value(%);
```

$$\int_0^{1/2\,\pi} 2\cos(t)\sin(t)\sqrt{4\cos(t)^2+\sin(t)^2}\;dt = \frac{14}{9}$$

Ví dụ 3: Tích phân đường loại 2 : $I = \int_C (x^2 - 2xy)dx + (y^2 - 2xy)dy$

với C là đường parabol y = x^2 , -1 ≤ x ≤ 1.

Dùng công thức : I= $\int_a^b (x^2 - 2xy) + (y^2 - 2xy)y'\,dx$

```
restart:with(student):
int((x^2-2*x^3)+(x^4-2*x^3)*diff(x^2,x),x=-1..1);
```

$$\frac{-14}{15}$$

III. PHƯƠNG PHÁP ĐỔI BIẾN

changevar(s , f)
changevar(s , f , u)
changevar(t , g , v)

Trong đó:

s – một biểu thức dạng h(x) = g(u), với x là hàm của biến u
f – một biểu thức tích phân dạng: Int(F(x), x = a...b)
u – biến tích phân mới.

t – Tập hợp các phương trình đổi biến.
g – là tích phân kép, tích phân bội ba ...
v – Danh sách biến mới.

Ví dụ 1: Tính bằng phương pháp đổi biến : $\int_0^a \sqrt{a^2 - x^2}\,dx$

```
with(student):
changevar(x=a*sin(t),Int(sqrt(a^2-x^2),x=0..a),t);
```

$$\int_0^{1/2\,\pi} \sqrt{a^2 - a^2 \sin(t)^2}\; a\cos(t)\,dt$$

```
%=value(%);
```

$$\int_0^{1/2\,\pi} \sqrt{a^2 - a^2 \sin(t)^2}\; a\cos(t)\,dt = \frac{1}{4}\sqrt{a^2}\; a\,\pi$$

Ví dụ 2: Tính $\int_0^1 \frac{x}{\sqrt{x+1}}\,dx$

```
changevar(t=sqrt(x+1),Int(x/sqrt(x+1),x=0..1),t);
```

$$\int_1^{\sqrt{2}} -2 + 2\,t^2\,dt$$

```
Int(x/sqrt(x+1),x=0..1)=value(%);
```

$$\int_0^1 \frac{x}{\sqrt{x+1}}\,dx = -\frac{2}{3}\sqrt{2} + \frac{4}{3}$$

Ví dụ 3 : Đổi sang tọa độ cực tích phân : $\iint\limits_D \frac{1}{\sqrt{4-x^2-y^2}}dxdy$

```
changevar({x=r*cos(phi),y=r*sin(phi)},
Doubleint(1/sqrt(4-x^2-y^2),x,y),[r,phi]);
```

$$\int\int \frac{|r|}{\sqrt{4-r^2}}\,dr\,d\phi$$

Ví dụ 4 : Đổi sang tọa độ cầu tích phân $\iiint \sqrt{x^2+y^2+z^2}\,dxdydz$

```
changevar({x=p*sin(theta)*cos(phi),
           y=p*sin(theta)*sin(phi),
           z=p*cos(theta)},
           Tripleint(sqrt(x^2+y^2+z^2),x,y,z),
           [p,theta,phi]);
```

$$\int\int\int |p|^3\,|\sin(\theta)|\,dp\,d\theta\,d\phi$$

IV. PHƯƠNG PHÁP TÍCH PHÂN TỪNG PHẦN

intparts(f, u)

trong đó :

f : một biểu thức của dạng **Int(u*dv, x)**

u : thừa số được vi phân.

Ví dụ 1: Tính tích phân $\int_0^{\frac{\pi}{2}} x\cos x dx$

```
with(student):
intparts(Int(x*cos(x),x=0..Pi/2),x);
```

$$\frac{1}{2}\pi - \int_0^{1/2\,\pi} \sin(x)\,dx$$

```
Int(x*cos(x),x=0..Pi/2)=value(%);
```

$$\int_0^{1/2\,\pi} x\cos(x)\,dx = \frac{1}{2}\pi - 1$$

Ví dụ 2: Tìm công thức truy chứng của $F_n = \int_0^1 x^n e^x dx$

```
Fn:= Int( x^n*exp(x), x=0..1 );
```

$$Fn := \int_0^1 x^n\, \mathrm{e}^x\, dx$$

```
simplify( intparts( Fn,x^n ) );
```

$$\mathrm{e} - n\int_0^1 x^{(n-1)}\,\mathrm{e}^x\,dx$$

Ta được công thức truy chứng : $F_n = e - n F_{n-1}$

Giải phương trình đệ qui nầy ta được công thức tính Fn

```
F(n)= exp(1)-n*F(n-1):
rsolve({%,F(0)=int(exp(x),x=0..1)},{F});
```

$$\{F(n) = -(-1)^{(-n)}\,\Gamma(n+1) + (-1)^{(-n)}\,\Gamma(n+1,-1)\}$$

Thực hành

1. Tính các tích phân bất định sau đây:

   a) $\int \sqrt{e^x - 1}\,dx$ b) $\int \frac{x}{\sqrt{(2ax - x^2)^3}}\,dx$

   c) $\int \sqrt{x^2 - a^2}\,dx$ d) $\int \frac{1}{x\sqrt{1+x^2}}\,dx$

   e) $\int \sec^2 x\,dx$ f) $\int \frac{1}{1+\sin x+\cos x}\,dx$

2. Tính $\int_0^1 \left(\int_0^1 \frac{x-y}{(x+y)^3}\,dy \right) dx$ và $\int_0^1 \left(\int_0^1 \frac{x-y}{(x+y)^3}\,dx \right) dy$.

   So sánh hai kết quả.

3. Tính các tích phân sau :

   a) $\int_0^{10} \frac{4x^4 + 4x^3 - 2x^2 - 10x + 6}{x^5 + 7x^4 + 16x^3 + 10x^2}\,dx$ b) $\int_0^{\pi/2} x^4 \sin x \cos x\,dx$

   c) $\int_{1/7}^{1/5} \frac{1}{x\sqrt{5x^2 - 6x + 1}}\,dx$ d) $\int_0^1 \frac{1}{\sqrt{1-x^2}}\,dx$

   e) $\int_0^{+\infty} e^{-ax}\cos^2(bx)\,dx$, $a > 0$ f) $\int_0^{+\infty} \frac{\sin x}{x}\,dx$

4. Cho C là miền $\{(x,y) \in |R^2 \,/\, 1/2 \le xy \le 2,\ 1 \le x \le 3\}$.

   Hãy tính $\iint_C \frac{e^{\frac{1}{xy}}}{y^2(x+1)^2}\,dxdy$

5. Tính các tích phân sau:

a) $\displaystyle\int \frac{\ln(x^2+1)}{x^2+1}dx$ b) $\displaystyle\int \frac{x^2}{\sqrt{x^6+1}}dx$

c) $\displaystyle\int_0^{+\infty} \frac{\ln x}{(x+a)(x-1)}dx$ d) $\displaystyle\int \mathrm{tg}\left(\frac{1}{3}arctgx\right)dx$

e) $\displaystyle\int \arcsin^2\left(\frac{x}{a}\right)dx$ f) $\displaystyle\int \frac{1}{\sqrt{a\sqrt{x}+b}}dx$

6. Tính diện tích hình phẳng giới hạn bởi :
   a) $y^2 = 2x+1$ và $x - y - 1 = 0$
   b) đường cong kín $y^2 = x^2 - y^2$
   c) $y = x^3 - 5x^2 + 2x + 7$ và $y = 2x - 1$

7. Tính thể tích vật thể tròn xoay tạo từ hình phẳng giới hạn bởi $y^2 = x^3$ và $x=1$ quay quanh ox.

8. Tính thể tích vật thể giới hạn bởi các mặt có phương trình :
   a) $y = x^2$; $z = y$; $z + y = 2$
   b) $x^2 + y^2$; $z = -2$; $y + z = 2$
   c) $x^2 + y^2 + z^2 = 4$; $x^2 + y^2 = 1$
   d) $z = \dfrac{\sqrt{x^2+y^2}}{4}$; $z = 0$; $x^2 + y^2 = 2$

9. Tìm các công thức truy chứng của :
   a) $F_n = \displaystyle\int_0^{\frac{\pi}{2}} \cos^n x.dx$
   b) $Tn = \displaystyle\int_0^{\frac{\pi}{4}} tg^n xdx$

CHUỖI SỐ – CHUỖI HÀM

1. TỔNG RIÊNG – CHUỖI SỐ

- **sum(f,k);** Tính tổng của f(k) theo công thức g(k) thỏa: g(k+1)-g(k)=f(k) mọi k.
- **sum(f,k=m..n);** Tính tổng: f(m) + f(m+1) + ... + f(n)
- **sum(f,k=alpha);** Tính tổng biểu thức f(k) với k = alpha là nghiệm của đa thức, thường dùng lệnh RootOf

Ví dụ1: Tính $\sum_{k=0}^{n-1} k^2$

```
sum(k^2,k=0..n-1);
```

$$\frac{1}{3}n^3 - \frac{1}{2}n^2 + \frac{1}{6}n$$

```
factor(%);
```

$$\frac{1}{6}n\,(2\,n-1)\,(n-1)$$

Ví dụ 2: Tính $\sum_{k=1}^{\infty} \frac{1}{k^2}$

```
sum(1/(k^2),k=1..infinity);
```

$$\frac{1}{6}\pi^2$$

Ví dụ 3: Tính tổng $\frac{k}{k+1}$ với k là các nghiệm phương trình : $x^3 - 2 = 0$

```
sum(k/(k+1), k=RootOf(x^3-2));
```

2

Trong một số trường hợp, tổng được biểu diễn qua hàm:

$$\Psi(x) = \frac{\partial}{\partial x}\left(\ln(\Gamma(x))\right) \quad \text{hay } \Psi(n,x) = \frac{\partial^n}{\partial x^n}\left(\ln(\Gamma(x))\right) \quad \text{với } \Gamma(x) = \int_0^{+\infty} e^{-t} t^{x-1} dt$$

như ví dụ sau đây:

Sum(k/(k+1),k=0..n) = sum(k/(k+1), k=0..n);

$$\sum_{k=0}^{n} \frac{k}{k+1} = n + 1 - \Psi(n+2) - \gamma$$

2. CHUỖI TAYLOR

- **series(expr, eqn)**
- **series(expr, eqn, n)**

trong đó:

expr – một biểu thức hay hàm theo x cần khai triển thành chuỗi.
eqn – tên biến x hoặc phương trình dạng x=a để khai triển tại 0 hay a
n – (tùy chọn) là số nguyên không âm n, khai triển đến cấp n, giá trị ngầm định n=6.

Ví dụ 1: Khai triển Taylor của hàm số f(x) = $\frac{1}{x}$ tại x=1.

```
series(1/x,x=1);
```

$$1-(x-1)+(x-1)^2-(x-1)^3+(x-1)^4-(x-1)^5+O((x-1)^6)$$

Ví dụ 2: Khai triển Mac-Laurin của hàm số f(x) = cosx đến cấp 10.

```
series(cos(x),x,10);
```

$$1-\frac{1}{2}x^2+\frac{1}{24}x^4-\frac{1}{720}x^6+\frac{1}{40320}x^8+O(x^{10})$$

Ví dụ 3 : Tính gần đúng $\sqrt{e}$ tới số hạng thứ 6.

```
series(exp(x),x,6);
```

$$1 + x + \frac{1}{2}x^2 + \frac{1}{6}x^3 + \frac{1}{24}x^4 + \frac{1}{120}x^5 + O(x^6)$$

```
eval(%,x=0.5);
```

1.648697917

Thực hành

1. Tính tổng của các chuỗi số :

a). $\sum_{n=3}^{\infty} \frac{1}{n^2 - 4}$ b). $\sum_{n=1}^{\infty} \frac{1}{n^2 + 3n + 2}$

c). $\sum_{n=0}^{\infty} \frac{2 + (-1)^n}{3^n}$ d). $\sum_{n=0}^{\infty} \operatorname{arctg}\left(\frac{1}{n^2 + n + 1}\right)$

2. Xét sự hội tụ của các chuỗi số :

a). $\sum_{n=1}^{\infty} \frac{n+1}{n^2 + 2}$ b). $\sum_{n=1}^{\infty} \frac{1}{\sqrt{n(n^2+1)}}$

c). $\sum_{n=1}^{\infty} \left(\frac{2n+3}{n\sqrt{n} + n + 1}\right)^2$ d). $\sum_{n=1}^{\infty} \frac{2^n - 1}{n2^{n+1} + 2}$

e). $\sum_{n=1}^{\infty} \frac{\sqrt[n]{e} - 1}{n}$ f). $\sum_{n=2}^{\infty} \frac{\ln\sqrt{n}}{n}$

g). $1 + \frac{1}{3} + \frac{1}{5} + \frac{1}{7} + ...$ h). $\frac{1}{1.3} + \frac{2}{3.5} + \frac{3}{5.7} + \frac{4}{7.9} + ...$

3. Viết khai triển Taylor của hàm $f(x) = \frac{1}{1-x}$ tại lân cận x = a

4. Viết khai triển Taylor của hàm $f(x) = \sqrt[3]{x}$ tại lân cận x=1

5. Viết khai triển Mac-Laurin của các hàm số sau :

a). $f(x) = \frac{1}{(x-1)^2}$ b). $f(x) = \cos 2x$ c). $f(x) = \frac{x}{1+x^2}$

6. Viết khai triển Mac-Laurin của hàm số $f(x) = \int_0^x \frac{\sin t}{t}\,dt$,

tính gần đúng tích phân $\int_0^{\frac{1}{2}} \frac{\sin t}{t}\,dt$ với sai số không vượt quá 10^{-3}

I. PHƯƠNG TRÌNH VI PHÂN

Câu lệnh : **dsolve(ODE);**
dsolve({ODE, ICs}, y(x), Options);

trong đó :
- ODE - phương trình vi phân thường (an ordinary differential equation)
- y(x) - hàm 1 biến độc lập
- Options - tùy chọn bao gồm: implicit (dạng ẩn) , explicit (dạng hiện –ngầm định) , useInt (Dạng tích phân) , series (chuỗi) , numeric (dạng số)...
- ICs - Điều kiện ban đầu (để tìm nghiệm riêng)
  dạng: y(a) = b , D(y)(a) = c , D(D(y))(a) = d...

Ví dụ 1: Giải phương trình vi phân : y'= y.tgx

```
ode:=diff(y(x),x)=y(x)*tan(x);
```

$$ode := \frac{\partial}{\partial x}\mathrm{y}(x) = \mathrm{y}(x)\tan(x)$$

dsolve(ode,y(x)); $\mathrm{y}(x) = \frac{\_C1}{\cos(x)}$

trong đó _C1 là hằng số tùy ý

Để giản tiện khi nhập phương trình vi phân, ta có thể dùng lệnh alias như ví dụ sau:

Ví dụ 2 : Giải phương trình vi phân : y' = $\frac{2xy}{x^2 - y^2}$

```
alias(y=y(x)):
ptvf:=diff(y,x)=(2*x*y)/(x^2-y^2);
```

$$ptvf := \frac{\partial}{\partial x}y = 2\frac{x\,y}{x^2 - y^2}$$

```
dsolve(ptvf,y);
```

$$y = \frac{1}{2}\frac{1 - \sqrt{1 - 4\,\_C1^2\,x^2}}{\_C1},\ y = \frac{1}{2}\frac{1 + \sqrt{1 - 4\,\_C1^2\,x^2}}{\_C1}$$

Ví dụ 3 : Giải phương trình vi phân : y' = $\frac{y}{x + y^3}$

```
ode:=diff(y(x),x)=y(x)/(x+y(x)^3);
```

$$ode := \frac{\partial}{\partial x}\mathrm{y}(x) = \frac{\mathrm{y}(x)}{x + \mathrm{y}(x)^3}$$

```
dsolve(ode,useInt);
```

$$x - \left(\int^{\mathrm{y}(x)} \_b^2\, \mathbf{e}^{\left(-\int^{\_b} \frac{1}{\_b}\, d\_b\right)}\, d\_b + \_C1\right) \mathbf{e}^{\left(\int^{\mathrm{y}(x)} \frac{1}{\_a}\, d\_a\right)} = 0$$

```
value(%);
```

$$x - \left(\frac{1}{2}\mathrm{y}(x)^2 + \_C1\right)\mathrm{y}(x) = 0$$

Ví dụ 4 : Giải phương trình vi phân : xy''= y' $\ln(\frac{y'}{x})$

```
ode:=x*diff(y(x),x,x)=diff(y(x),x)*ln(diff(y(x),x)/x);
```

$$ode := x\left(\frac{\partial^2}{\partial x^2}\mathrm{y}(x)\right) = \left(\frac{\partial}{\partial x}\mathrm{y}(x)\right)\ln\left(\frac{\frac{\partial}{\partial x}\mathrm{y}(x)}{x}\right)$$

```
dsolve(ode);
```

$$\mathrm{y}(x) = -\frac{\mathbf{e}\,\mathbf{e}^{(\_C1\,x)}}{\_C1^2} + \frac{\mathbf{e}\,\mathbf{e}^{(\_C1\,x)}\,x}{\_C1} + \_C2$$

Ví dụ 5 : Giải phương trình vi phân : y'' - $\frac{y'}{x-1}$ = x(x-1)

thỏa điều kiện : y(2)=1, y'(2)= -1

```
ode:=diff(y(x),x$2)-diff(y(x),x)/(x-1) =x*(x-1);
```

$$ode := \left(\frac{\partial^2}{\partial x^2}\mathrm{y}(x)\right) - \frac{\frac{\partial}{\partial x}\mathrm{y}(x)}{x-1} = x\,(x-1)$$

```
dsolve({ode,y(2)=1,D(y)(2)=-1},y(x));
```

$$\mathrm{y}(x) = \frac{1}{8}x^4 - \frac{1}{6}x^3 - \frac{3}{2}x^2 + 3\,x + \frac{1}{3}$$

Ví dụ 6: Tìm nghiệm riêng của phương trình vi phân : yy''- y'^{2} = 0 thỏa : y(0) = 1, y'(0) = 2

```
ode:=y(x)*diff(y(x),x$2)-diff(y(x),x)^2=0;
```

$$ode := \mathrm{y}(x)\left(\frac{\partial^2}{\partial x^2}\mathrm{y}(x)\right) - \left(\frac{\partial}{\partial x}\mathrm{y}(x)\right)^2 = 0$$

```
dsolve({ode,y(0)=1,D(y)(0)=2});
```

$$\mathrm{y}(x) = \mathbf{e}^{(2\,x)}$$

Ví dụ 7: Giải phương trình vi phân : xy''+y'+4x^{2}y = 0

```
alias(y=y(x)):
ode:=x*diff(y,x$2)+diff(y,x)+4*x^2*y=0;
```

$$ode := x\left(\frac{\partial^2}{\partial x^2}y\right) + \left(\frac{\partial}{\partial x}y\right) + 4\,x^2\,y = 0$$

```
dsolve(ode,y);
```

$$y = \_C1\ \mathrm{BesselJ}\left(0, \frac{4}{3}x^{(3/2)}\right) + \_C2\ \mathrm{BesselY}\left(0, \frac{4}{3}x^{(3/2)}\right)$$

```
dsolve(ode,y,series);
```

$$y = \_C1\left(1 - \frac{4}{9}x^3 + \mathrm{O}(x^6)\right) + \_C2\left(\ln(x)\left(1 - \frac{4}{9}x^3 + \mathrm{O}(x^6)\right) + \left(\frac{8}{27}x^3 + \mathrm{O}(x^6)\right)\right)$$

Để giải gần đúng phương trình vi phân, ta dùng lệnh:

dsolve({ode, init} , vars, numeric)

dsolve({ode, init}, vars, numeric, options)

trong đó :

{ode, init} - tập hợp các phương trình vi phân và điều kiện ban đầu.

vars - các hàm cần tìm.

options - các lựa chọn dạng: keyword=value như

- method=rkf45 (Fehlberg fourth-fifth order Runge-Kutta method), method=dverk78, method=classical, method=gear, method=mgear, method=lsode, method=taylorseries. Ngầm định là method=rkf45
- value = xarray dạng : value=array([x_1,x_2,...,x_k])

Ví dụ 8 : Giải gần đúng phương trình vi phân : y' = -x +y

thỏa y(0) = 1,5

```
alias(y=y(x),y0=y(0)):
ode:=diff(y,x)=y-x;
```

$$ode := \frac{\partial}{\partial x} y = y - x$$

Tính gần đúng với các giá trị y(x) với x= 0 ; 0.25 ; 0.5 ; 0.75 ; 1

```
F:=dsolve({ode,y0=1.5},y,'numeric',
          value=array([0,0.25,0.5,0.75,1]));
```

$$F := \left[\begin{array}{c} [x, y] \\ \left[\begin{array}{cc} 0 & 1.5 \\ .25 & 1.892012706 \\ .5 & 2.324360630 \\ .75 & 2.808499999 \\ 1 & 3.359140899 \end{array}\right] \end{array}\right]$$

Ví dụ 9: Giải gần đúng phương trình vi phân : y''- (1-y^2) y'+ y = 0 thỏa y(0) = 0 ,y'(0) = -0,1

```
alias(y=y(t),y0=y(0),yp0=D(y)(0)):
ode:=diff(y,t$2)-(1-y^2)*diff(y,t)+y=0;
```

$$ode := \left(\frac{\partial^2}{\partial t^2} y\right) - (1 - y^2)\left(\frac{\partial}{\partial t} y\right) + y = 0$$

```
init:=y0=0,yp0=-0.1:
F:=dsolve({ode,init},y,'numeric');
```

$F :=$ proc $(rkf45\_x)$ ... end proc

Ta có thể vẽ đồ thị của hàm y theo t bằng các định nghĩa hàm Y(t) như sau :

```
Y:=t->rhs(op(1,select(hastype,F(t),'function')));
```

$$Y := t \rightarrow \text{rhs}(\text{op}(1, \text{select}(hastype, \text{F}(t), '\mathit{function}')))$$

```
plot(Y,0..14);
```



II. HỆ PHƯƠNG TRÌNH VI PHÂN

Câu lệnh : **dsolve({sysODE}, {funcs})**
dsolve({sysODE}, {funcs}, 'explicit')

trong đó :

{sysODE} - *tập hợp các phương trình vi phân được ngăn cách bởi dấu phẩy*
{funcs} - *tập hợp các hàm số cần xác định hoặc tên hàm.*
'explicit' - *tùy chọn để yêu cầu sắp xếp sự xuất hiện tập các nghiệm trong trường hợp hệ phương trình vi phân phi tuyến*

Ví dụ 1: Giải hệ phương trình vi phân : $\begin{cases} \dfrac{dy}{dx} = z \\ \dfrac{dz}{dx} = y \end{cases}$

```
sys1:={diff(y(x),x)=z(x),diff(z(x),x)=y(x)};
```

$$sys1 := \{ \frac{\partial}{\partial x} y(x) = z(x), \frac{\partial}{\partial x} z(x) = y(x) \}$$

```
dsolve(sys1,{y,z});
```

$$\{ y(x) = \_C1\ e^x - \_C2\ e^{(-x)}, z(x) = \_C1\ e^x + \_C2\ e^{(-x)} \}$$

Ví dụ 2 : Giải hệ phương trình vi phân $\begin{cases} \dfrac{dx}{dt} = x^2 + xy \\ \dfrac{dy}{dt} = xy + y^2 \end{cases}$

```
sys2:={diff(x(t),t)=x(t)^2+x(t)*y(t),
        diff(y(t),t)=x(t)*y(t)+y(t)^2}:
```

```
dsolve(sys2,{x,y});
```

$$\left[\{ y(t) = 0 \}, \{ x(t) = \frac{1}{-t + \_C1} \} \right],$$
$$\left[\{ y(t) = -\frac{1}{\_C1\ t + \_C2} \}, \left\{ x(t) = -\frac{-\left(\frac{\partial}{\partial t} y(t)\right) + y(t)^2}{y(t)} \right\} \right]$$

```
dsolve(sys2,{x,y},explicit);
```

$$\{ y(t) = 0, x(t) = \frac{1}{-t + \_C1} \}, \{$$
$$x(t) = \left(-\frac{\_C1}{(\_C1\ t + \_C2)^2} + \frac{1}{(\_C1\ t + \_C2)^2} \right) (\_C1\ t + \_C2),$$
$$y(t) = -\frac{1}{\_C1\ t + \_C2} \}$$

Thực hành

1. Giải các phương trình vi phân bằng Maple. Thử giải bằng nhiều phương pháp, tìm lời giải ở dạng đơn giản nhất và thử kiểm tra lại :

   $a).\ 3y^2y' + 16x = 12xy^3 \qquad b).\ xy' - y = x\,tg\left(\dfrac{y}{x}\right)$

2. Giải các phương vi phân sau:

   a) $y' + 2xy = xe^{-x^2}$ b) $(1+x^2)y' = 2xy + (1+x^2)^2$

   c) $(x+y-1)^2\,dy = 2(y+2)^2\,dx$ d) $y' - tg\left(\dfrac{y-2x}{x+1}\right) = \dfrac{y+2}{x+1}$

3. Giải các phương trình vi phân

   a) $y'' + y'tgx = \sin 2x$ b) $xy'' - y' - x\sin\left(\dfrac{y'}{x}\right) = 0$

   c) $y'' + 4y = \dfrac{1}{\sin^2 x}$ e) $y'' + 3y' + 2y = \dfrac{1}{e^x + 1}$

4. Chứng minh rằng hàm số $y = x\int_1^x e^{t^2}\,dt$ là một nghiệm của phương trình $xy' - y = x^2e^{x^2}$.

   Tìm nghiệm tổng quát của phương trình

5. Giải các hệ phương trình sau

   $a)\begin{cases} \dfrac{dx}{dt} = y - z \\ \dfrac{dy}{dt} = z - x \\ \dfrac{dz}{dt} = x - y \end{cases} \qquad b)\ \dfrac{dt}{xy} = \dfrac{dx}{ty} = \dfrac{dy}{tx} \qquad c)\begin{cases} y' = 5y + 4z \\ z' = 4y + 5z \end{cases}$

ĐẠI SỐ TUYẾN TÍNH - MA TRẬN

Maple 6.0 cung cấp trên 100 câu lệnh để giải quyết các vấn đề của Đại số tuyến tính được đóng gói trong linalg package và cần phải nạp bởi lệnh **with(linalg):** trước khi thực hiện tính toán. Linalg package bao gồm một số lệnh sau:

GramSchmidt	*JordanBlock*	*LUdecomp*	*QRdecomp*	*addcol*
addrow	*adjoint*	*angle*	*augment*	*backsub*
band	*basis*	*bezout*	*blockmatrix*	*charmat*
charpoly	*cholesky*	*col*	*coldim*	*colspace*
colspan	*companion*	*cond*	*copyinto*	*crossprod*
curl	*definite*	*delcols*	*delrows*	*det*
diag	*diverge*	*dotprod*	*eigenvalues*	*eigenvectors*
entermatrix	*equal*	*exponential*	*extend*	*ffgausselim*
fibonacci	*forwardsub*	*frobenius*	*gausselim*	*gaussjord*
geneqns	*genmatrix*	*grad*	*hadamard*	*hermite*
hessian	*hilbert*	*htranspose*	*ihermite*	*indexfunc*
innerprod	*intbasis*	*inverse*	*ismith*	*issimilar*
iszero	*jacobian*	*jordan*	*kernel*	*laplacian*
leastsqrs	*linsolve*	*matadd*	*matrix*	*minor*
minpoly	*mulcol*	*multiply*	*norm*	*normalize*
orthog	*permanent*	*pivot*	*potential*	*randmatrix*
randvector	*rank*	*references*	*row*	*rowdim*
rowspace	*rowspan*	*scalarmul*	*singularvals*	*smith*
stackmatrix	*submatrix*	*subvector*	*sumbasis*	*swapcol*
swaprow	*sylvester*	*toeplitz*	*trace*	*transpose*
vandermonde	*vecpotent*	*vectdim*	*vector*	*wronskian*

I. MA TRẬN-ĐỊNH THỨC

1. TẠO MA TRẬN

Câu lệnh : **matrix(L)**
matrix(m, n)
matrix(m, n, L)
matrix(m, n, f)

trong đó :

m,n - số nguyên dương chỉ số hàng và số cột.
L - Danh sách hoặc vectơ.
f - một hàm số dùng để tạo các phần tử của ma trận.

Ví dụ 1:

```
with(linalg):
A:=matrix(2,2,[1,2,3,4]);
```

$$A := \begin{bmatrix} 1 & 2 \\ 3 & 4 \end{bmatrix}$$

```
B:=matrix([[a,b],[c,d]]);
```

$$B := \begin{bmatrix} a & b \\ c & d \end{bmatrix}$$

```
C:=matrix(3,2,(i,j)->i+j-1);
```

$$C := \begin{bmatrix} 1 & 2 \\ 2 & 3 \\ 3 & 4 \end{bmatrix}$$

```
M:=matrix(2,2,0);
```

$$M := \begin{bmatrix} 0 & 0 \\ 0 & 0 \end{bmatrix}$$

```
N:=matrix(2,2,3);
```

$$N := \begin{bmatrix} 3 & 3 \\ 3 & 3 \end{bmatrix}$$

□ MA TRẬN CHUYỂN VỊ : **transpose(M);**

Ví dụ : `M:=matrix([[1,2,3],[x,y,z]]);` $M := \begin{bmatrix} 1 & 2 & 3 \\ x & y & z \end{bmatrix}$

`M^t=transpose(M);` $M^t = \begin{bmatrix} 1 & x \\ 2 & y \\ 3 & z \end{bmatrix}$

□ MA TRẬN ĐƠN VỊ , MA TRẬN ĐƯỜNG CHÉO:

Câu lệnh : **diag(B1, B2, .., Bn)** trong đó :
B1, B2, ..., Bn - là các ma trận vuông.

Ví dụ 2: Ma trận đơn vị , ma trận đường chéo

`I2=diag(1,1);` $I2 = \begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}$

`I3=diag(1,1,1);` $I3 = \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$

`m:=diag(lambda1, lambda2);` $m := \begin{bmatrix} \lambda 1 & 0 \\ 0 & \lambda 2 \end{bmatrix}$

`N:=diag(m,diag(1,1));` $N := \begin{bmatrix} \lambda 1 & 0 & 0 & 0 \\ 0 & \lambda 2 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$

□ MA TRẬN CON : **minor(A, r, c)**

trong đó : A–ma trận và r, c – bỏ đi hàng r và cột c của A

Ví dụ:

```
A := matrix( [[1,-6,1],[-6,1,1],[-1,6,1]] );
```

$A := \begin{bmatrix} 1 & -6 & 1 \\ -6 & 1 & 1 \\ -1 & 6 & 1 \end{bmatrix}$

`A11:=minor(A,1,1);` $A11 := \begin{bmatrix} 1 & 1 \\ 6 & 1 \end{bmatrix}$

`A21:=minor(A,2,1);` $A21 := \begin{bmatrix} -6 & 1 \\ 6 & 1 \end{bmatrix}$

□ MA TRẬN KHỐI: **blockmatrix(m,n,L);**

trong đó : m,n : số hàng , số cột của ma trận khối mà các phần tử là ma trận trong danh sách L
L - danh sách gồm m*n các ma trận.

`A := matrix(2,2,(i,j) -> abs(i-j));` $A := \begin{bmatrix} 0 & 1 \\ 1 & 0 \end{bmatrix}$

`B:=matrix(2,2,[x,x,x,x]);` $B := \begin{bmatrix} x & x \\ x & x \end{bmatrix}$

`blockmatrix(1,2,[A,B]);` $\begin{bmatrix} 0 & 1 & x & x \\ 1 & 0 & x & x \end{bmatrix}$

`blockmatrix(2,2,[A,B,B,A]);` $\begin{bmatrix} 0 & 1 & x & x \\ 1 & 0 & x & x \\ x & x & 0 & 1 \\ x & x & 1 & 0 \end{bmatrix}$

2. PHÉP TÍNH MA TRẬN

Các phép toán cộng, trừ hai ma trận cùng cấp, nhân một số với ma trận, nhân hai ma trận, lũy thừa một ma trận ta dùng các ký hiệu : + , - , * , &*, ^ và lệnh **evalm(M);** để đánh giá kết quả của ma trận M.

Ví dụ: Cho ma trận A=$\begin{bmatrix}1 & -2\\ 3 & 4\end{bmatrix}$ và B=$\begin{bmatrix}\alpha & \beta\\ \gamma & \delta\end{bmatrix}$.

Tính 2A-B ; AB , A^3 , A^{-1}.

```
with(linalg):
A:=matrix(2,2,[1,-2,3,4]);
```

$$A := \begin{bmatrix}1 & -2\\ 3 & 4\end{bmatrix}$$

```
B:=matrix([[alpha,beta],[gamma,delta]]);
```

$$B := \begin{bmatrix}\alpha & \beta\\ \gamma & \delta\end{bmatrix}$$

```
2*A-B=evalm(2*A-B);
```

$$2\,A - B = \begin{bmatrix}2-\alpha & -4-\beta\\ 6-\gamma & 8-\delta\end{bmatrix}$$

```
A*B=evalm(A&*B);
```

$$A\,B = \begin{bmatrix}\alpha - 2\,\gamma & \beta - 2\,\delta\\ 3\,\alpha + 4\,\gamma & 3\,\beta + 4\,\delta\end{bmatrix}$$

```
A^3=evalm(A^3);
```

$$A^3 = \begin{bmatrix}-35 & -30\\ 45 & 10\end{bmatrix}$$

```
1/A=evalm(1/A);
```

$$\frac{1}{A} = \begin{bmatrix}\frac{2}{5} & \frac{1}{5}\\ \frac{-3}{10} & \frac{1}{10}\end{bmatrix}$$

Đối với phép cộng, trừ thì $A \pm n$ *là* $\pm$ *các phần tử* A_{ii} *bởi số n.*

```
evalm(B-3);
```

$$\begin{bmatrix}\alpha - 3 & \beta\\ \gamma & \delta - 3\end{bmatrix}$$

Trong trường hợp để thực hiện các thủ tục (procedure) trên tất cả các phần tử của ma trận, ta thường dùng câu lệnh:

map(fcn, expr, arg2, ..., argn)

trong đó :

fcn – thủ tục cần thực hiện trên các số hạng của biểu thức expr

expr – biểu thức , ma trận ...

- Thừa số các số hạng của ma trận M : **map(factor,M);**

Ví dụ :

```
A:=toeplitz([alpha,beta]);
```

$$A := \begin{bmatrix}\alpha & \beta\\ \beta & \alpha\end{bmatrix}$$

```
evalm(A^3);
```

$$\begin{bmatrix}(\alpha^2+\beta^2)\,\alpha + 2\,\alpha\,\beta^2 & (\alpha^2+\beta^2)\,\beta + 2\,\alpha^2\,\beta\\ (\alpha^2+\beta^2)\,\beta + 2\,\alpha^2\,\beta & (\alpha^2+\beta^2)\,\alpha + 2\,\alpha\,\beta^2\end{bmatrix}$$

```
map(factor,%);
```

$$\begin{bmatrix} \alpha\,(\alpha^2+3\,\beta^2) & \beta\,(3\,\alpha^2+\beta^2) \\ \beta\,(3\,\alpha^2+\beta^2) & \alpha\,(\alpha^2+3\,\beta^2) \end{bmatrix}$$

□ Đánh giá đầy đủ (full evaluation) :

```
R:=matrix(2,2,[cos(alpha),
      -sin(alpha),sin(alpha),cos(alpha)]);
```

$$R := \begin{bmatrix} \cos(\alpha) & -\sin(\alpha) \\ \sin(\alpha) & \cos(\alpha) \end{bmatrix}$$

ta gán giá trị alpha=1

```
alpha:=1:
```

lệnh *evalm* không đánh giá đầy đủ giá trị ma trận R

```
evalm(R);
```

$$\begin{bmatrix} \cos(\alpha) & -\sin(\alpha) \\ \sin(\alpha) & \cos(\alpha) \end{bmatrix}$$

nhưng lệnh *map* đánh giá đầy đủ theo các phép gán

```
map(eval,R);
```

$$\begin{bmatrix} \cos(1) & -\sin(1) \\ \sin(1) & \cos(1) \end{bmatrix}$$

□ Đạo hàm các phần tử ma trận:

```
R:=matrix(2,2,[cos(x^2),-sin(x),sin(x),cos(x^2)]);
```

$$R := \begin{bmatrix} \cos(x^2) & -\sin(x) \\ \sin(x) & \cos(x^2) \end{bmatrix}$$

```
map(diff,R,x);
```

$$\begin{bmatrix} -2\sin(x^2)\,x & -\cos(x) \\ \cos(x) & -2\sin(x^2)\,x \end{bmatrix}$$

3. RÚT GỌN MA TRẬN

<u>Câu lệnh</u>: **gaussjord(A)**
gaussjord(A, 'r')
gaussjord(A, 'r', 'd')

A - ma trận chữ nhật.
'r' - (tùy chọn) cho ra hạng ma trận A gán vào r.
'd' - (tùy chọn) cho ra định thức A gán vào d, nếu A vuông.

<u>Ví dụ</u>:

```
A:=array([[4,-6,1,0],[-6,12,0,1],[-2,6,1,1]]);
```

$$A := \begin{bmatrix} 4 & -6 & 1 & 0 \\ -6 & 12 & 0 & 1 \\ -2 & 6 & 1 & 1 \end{bmatrix}$$

```
gaussjord(A,'r');
```

$$\begin{bmatrix} 1 & 0 & 1 & \frac{1}{2} \\ 0 & 1 & \frac{1}{2} & \frac{1}{3} \\ 0 & 0 & 0 & 0 \end{bmatrix}$$

```
'rank'(A)=r;
```

$$\text{rank}(A) = 2$$

4. ĐỊNH THỨC

Câu lệnh: **det(M);**

Ví dụ:

```
A := array( [[4,-6,m],[-6,m,1],[-m,6,1]] );
```

$$A := \begin{bmatrix} 4 & -6 & m \\ -6 & m & 1 \\ -m & 6 & 1 \end{bmatrix}$$

```
det(A);
```

$$-26\,m - 60 + m^3$$

5. HẠNG MA TRẬN

Câu lệnh: **rank(M);**

Ví dụ:

```
A := matrix(3,3, [x,1,0,0,0,1,x*y,y,1]);
rank(A);
```

$$2$$

```
gaussjord(A);
```

$$\begin{bmatrix} 1 & \frac{1}{x} & 0 \\ 0 & 0 & 1 \\ 0 & 0 & 0 \end{bmatrix}$$

6. MA TRẬN NGHỊCH ĐẢO

- Tính bằng phép tính A^{-1}:

  Dùng lệnh **inverse(A);** hoặc **evalm(1/A);**

```
A := matrix( [[1,-6,1],[-6,1,1],[-1,6,1]] );
```

$$A := \begin{bmatrix} 1 & -6 & 1 \\ -6 & 1 & 1 \\ -1 & 6 & 1 \end{bmatrix}$$

```
inverse(A);
```

$$\begin{bmatrix} \frac{1}{14} & \frac{-6}{35} & \frac{1}{10} \\ \frac{-1}{14} & \frac{-1}{35} & \frac{1}{10} \\ \frac{1}{2} & 0 & \frac{1}{2} \end{bmatrix}$$

```
evalm(1/A);
```

$$\begin{bmatrix} \frac{1}{14} & \frac{-6}{35} & \frac{1}{10} \\ \frac{-1}{14} & \frac{-1}{35} & \frac{1}{10} \\ \frac{1}{2} & 0 & \frac{1}{2} \end{bmatrix}$$

- Tính bằng phép rút gọn ma trận: nối ma trận A và I bởi lệnh **blockmatrix** hoặc **augment(A,I)**

```
A := matrix( [[1,-6,1],[-6,1,1],[-1,6,1]] );
```

$$A := \begin{bmatrix} 1 & -6 & 1 \\ -6 & 1 & 1 \\ -1 & 6 & 1 \end{bmatrix}$$

```
AI:=blockmatrix(1,2,[A,diag(1,1,1)]);
```

$$AI := \begin{bmatrix} 1 & -6 & 1 & 1 & 0 & 0 \\ -6 & 1 & 1 & 0 & 1 & 0 \\ -1 & 6 & 1 & 0 & 0 & 1 \end{bmatrix}$$

```
gaussjord(AI);
```

$$\begin{bmatrix} 1 & 0 & 0 & \frac{1}{14} & \frac{-6}{35} & \frac{1}{10} \\ 0 & 1 & 0 & \frac{-1}{14} & \frac{-1}{35} & \frac{1}{10} \\ 0 & 0 & 1 & \frac{1}{2} & 0 & \frac{1}{2} \end{bmatrix}$$

- Tính bằng công thức: $A^{-1} = \frac{1}{\det(A)} P_A$, Tính P_A bởi lệnh **adjoint(A);** hoặc **adj(A);**

```
A := matrix( [[1,-6,1],[-6,1,1],[-1,6,1]] );
```

$$A := \begin{bmatrix} 1 & -6 & 1 \\ -6 & 1 & 1 \\ -1 & 6 & 1 \end{bmatrix}$$

```
PA:=adj(A);
```

$$PA := \begin{bmatrix} -5 & 12 & -7 \\ 5 & 2 & -7 \\ -35 & 0 & -35 \end{bmatrix}$$

```
evalm(PA/det(A));
```

$$\begin{bmatrix} \frac{1}{14} & \frac{-6}{35} & \frac{1}{10} \\ \frac{-1}{14} & \frac{-1}{35} & \frac{1}{10} \\ \frac{1}{2} & 0 & \frac{1}{2} \end{bmatrix}$$

II. HỆ PHƯƠNG TRÌNH TUYẾN TÍNH

Câu lệnh: **linsolve(A, B, 'r', v)**

trong đó:

A – là ma trận hệ số

B – là ma trận hoặc véctơ vế phải

r – (tùy chọn) là hạng của A

v – (tùy chọn) tên biến tự do, ngầm định là $\_t_1 , \_t_2$, ...

Ví dụ 1: Giải hệ phương trình : $\begin{cases} x + 2y = 1 \\ x + 3y = -2 \end{cases}$

```
A:= matrix( [[1,2],[1,3]] );
```

$$A := \begin{bmatrix} 1 & 2 \\ 1 & 3 \end{bmatrix}$$

```
B:=vector([1,-2]);
```

$$B := [1, -2]$$

```
X:=linsolve(A, B);
```

$$X := \begin{bmatrix} 7 \\ -3 \end{bmatrix}$$

Hệ có nghiệm duy nhất : x= 7 , y=-3

Ví dụ2: Giải hệ phương trình : $\begin{cases} x_1 - x_2 + x_3 - x_4 = 1 \\ x_1 - 2x_2 + x_3 + 3x_4 = 2 \\ 2x_1 + x_2 - x_3 - 2x_4 = 5 \\ 2x_1 + 2x_2 - x_3 - 6x_4 = 4 \end{cases}$

```
A:= matrix([[1,-1,1,-1],[1,-2,1,3],
            [2,1,-1,-2],[2,2,-1,-6]]):
B:= vector([1,2,5,4]):
```

```
X:=linsolve(A,B);
```
$X := [2 + \_t_1, -1 + 4\_t_1, -2 + 4\_t_1, \_t_1]$

Hệ có vô số nghiệm :

$x_1 = 2 + t$, $x_2 = -1+4t$, $x_3 = -2 +4t$, $x_4 = t$, (với t tùy ý)

Ví dụ 3: Giải hệ phương trình $\begin{cases} 2x - y + z = -2 \\ x + 2y + 3z = -1 \\ x - 3y - 2z = 3 \end{cases}$

```
A := matrix( [[2,-1,1],[1,2,3],[1,-3,-2]] ):
B:= vector([-2,-1,3]);
```
$B := [-2, -1, 3]$

```
X:=linsolve(A,B);
```
$X :=$

Hệ vô nghiệm

Ví dụ 4: Giải và biện luận theo m hệ phương trình : $\begin{cases} mx + y + z = 1 \\ x + my + z = m \\ x + y + mz = m^2 \end{cases}$

```
A:= matrix( [[m,1,1],[1,m,1],[1,1,m]] ):
B:= vector([1,m,m^2]):
m<>solve(det(A)=0,m);
```
$m \neq (-2, 1, 1)$

```
X:=linsolve(A,B);
```
$X := \left[-\frac{m+1}{m+2}, \frac{1}{m+2}, \frac{(m+1)^2}{m+2} \right]$

```
m:=1:
A1:=map(eval,A):B1:=map(eval,B):
X:=linsolve(A1,B1);
```
$X := [1 - \_t_1 - \_t_2, \_t_1, \_t_2]$

```
m:=-2:
A2:=map(eval,A):B2:=map(eval,B):
X:=linsolve(A2,B2);
```
$X :=$

III. KHÔNG GIAN VÉCTƠ

1. Tạo một véctơ

Dùng một trong các lệnh :

vector([x1, ..., xn])

vector(n)

vector(n, f)

trong đó :

x1, ..., xn - thành phần của vector có kiểu số.

n - số chiều của vector.

f - hàm số f được dùng tạo các thành phần của vector.

Do đó vector(n,f) tương đương với vector(1..n , [f(1),f(2) , …, f(n)])

Ví dụ :

```
vector( [5,4,6,3] );
```
$[5, 4, 6, 3]$

```
vector(4);
```
$[?_1, ?_2, ?_3, ?_4]$

```
vector(4, 0);
```
$[0, 0, 0, 0]$

Ví dụ: Tạo vector có chiều 4 bởi hàm số f(x) = x^2

```
f := x -> x^2:
v := vector(4, f);
```

$v := [1, 4, 9, 16]$

* Tạo vectơ cột có các thành phần theo công thức x^{j-1}

Ví dụ :

```
f:= (j) -> x^(j-1):
Vector(3,f)
```

$$\begin{bmatrix} 1 \\ x \\ x^2 \end{bmatrix}$$

Một số câu lệnh liên quan đến vector

Tên lệnh	Ý nghĩa
norm(v,2)	*Môđun của v*
normalize(v)	*Chuẩn hóa v*
scalarmul(v,k)	*Nhân vector v với số k*
dotprod(u,v)	*Tích vô hướng uv*
crossprod(u,v)	*Tích có hướng u^v*
GramSchmidt([$v_1,v_2,\ldots,v_n$])	*Trực giao hóa GramSchmidt các vector độc lập tuyến tính* $v_1,v_2,\ldots,v_n$
GramSchmidt([$v_1,v_2,\ldots,v_n$],normalized)	*Trực chuẩn hóa GramSchmidt các vector độc lập tuyến tính* $v_1,v_2,\ldots,v_n$
basis({$v_1,v_2,\ldots,v_n$})	*xác định cơ sở của không gian sinh bởi* $v_1,v_2,\ldots,v_n$
intbasis(V1,V2,…,Vn)	*xác định cơ sở không gian giao của các không gian sinh bởi hệ vector V1, V2,…,Vn*
sumbasis(V1,V2,…,Vn)	*xác định cơ sở không gian tổng của các không gian sinh bởi hệ vector V1, V2,…,Vn*
rowspace(A)	*xác định cơ sở không gian sinh bởi các hàng của ma trận A*
colspace(A)	*xác định cơ sở không gian sinh bởi các cột của ma trận A*
kernel(A) , nullspace(A)	*xác định cơ sở không gian nghiệm hệ phương trình thuần nhất AX=0*

2. Các ví dụ

- Tính môđun và chuẩn hóa vector v =(2,2,-1)

```
with(linalg):
v:=vector([2,2,-1]);
```

$v := [2, 2, -1]$

```
norm(v,2);
```

3

```
normalize(v);
```

$\left[\frac{2}{3}, \frac{2}{3}, \frac{-1}{3}\right]$

- Tính tích vô hướng và tích hữu hướng của u=(2,2,-1), v=(1,2,3)

```
v:=vector([2,2,-1]):
u:=vector([1,2,3]):
dotprod(u,v);
```

3

```
crossprod(u,v);
```

$[-8, 7, -2]$

- Xét tính độc lập tuyến tính của hệ vector

```
v1:=vector([1,2,-1,0]):
v2:=vector([2,1,2,3]) :
v3:=vector([-1,4,-7,-6]):
basis({v1,v2,v3});
```

$\{v2, v1\}$

$\Rightarrow$ Hệ phụ thuộc tuyến tính.

```
u1 := vector([2,2,2]):
u2 := vector([0,2,2]):
u3 := vector([0,0,2]):
basis({u1,u2,u3});
```
$\{u1, u2, u3\}$

$\Rightarrow$ Hệ độc lập tuyến tính.

- Trực giao hóa và trực chuẩn hóa hệ vector

```
u1 := vector([2,2,2]):
u2 := vector([0,2,2]):
u3 := vector([0,0,2]):
GramSchmidt([u1,u2,u3]);
```
$$\left[[2, 2, 2], \left[\frac{-4}{3}, \frac{2}{3}, \frac{2}{3}\right], [0, -1, 1]\right]$$

```
GramSchmidt([u1,u2,u3],normalized);
```
$$\left[\left[\frac{1}{3}\sqrt{3}, \frac{1}{3}\sqrt{3}, \frac{1}{3}\sqrt{3}\right], \left[-\frac{1}{6}\sqrt{8}\sqrt{3}, \frac{1}{12}\sqrt{8}\sqrt{3}, \frac{1}{12}\sqrt{8}\sqrt{3}\right], \left[0, -\frac{1}{2}\sqrt{2}, \frac{1}{2}\sqrt{2}\right]\right]$$

- Cho không gian sinh bởi : U = < (1,2,1) ; (1,0,2) > và W=<(2,-1,2) ; (2,1,1)>
  Tìm cơ sở của U, W, U$\cap$W và U+W

```
U:={vector([1,2,1]),vector([1,0,2])};
```
$U := \{[1, 2, 1], [1, 0, 2]\}$

```
W:={vector([2,-1,2]),vector([2,1,1])};
```
$W := \{[2, -1, 2], [2, 1, 1]\}$

```
basis(U);
```
$\{[1, 2, 1], [1, 0, 2]\}$

```
basis(W);
```
$\{[2, -1, 2], [2, 1, 1]\}$

```
intbasis(U,W);
```
$\{[0, 2, -1]\}$

```
sumbasis(U,W);
```
$\{[2, -1, 2], [1, 0, 2], [1, 2, 1]\}$

- Tìm cơ sở của không gian con
  U=$\{(x_1,x_2,x_3) / x_1+x_2+x_3= 0 , x_1+2x_2 -x_3= 0\}$
  A là ma trận hệ số của hệ phương trình :
  $x_1+x_2+x_3=0 , x_1+2x_2 -x_3= 0$

```
A:=matrix([[1,1,1],[1,2,-1]]);
```
$$A := \begin{bmatrix} 1 & 1 & 1 \\ 1 & 2 & -1 \end{bmatrix}$$

```
kernel(A);
```
$\left\{\left[\frac{-3}{2}, 1, \frac{1}{2}\right]\right\}$

3. Tọa độ vectơ – Chuyển cơ sở :

- Tọa độ vectơ :
  Để tính tọa độ vectơ v trong cơ sở B=$\{e_1,e_2,\ldots,e_n\}$ ta giải hệ phương trình tuyến tính : $B^TX=v$
  với B^T ma trận chuyển vị của ma trận B mà các hàng là các vectơ cơ sở $e_1,e_2,\ldots,e_n$

Ví dụ 1: Tìm tọa độ của v=(6,9,14) trong cơ sở
B={(1,1,1) , (1,1,2) , (1,2,3)}

```
with(linalg):
Bt:=transpose(matrix([[1,1,1],[1,1,2],[1,2,3]]));
```

$$Bt := \begin{bmatrix} 1 & 1 & 1 \\ 1 & 1 & 2 \\ 1 & 2 & 3 \end{bmatrix}$$

```
v:=vector([6,9,14]);
```
$v := [6, 9, 14]$

```
linsolve(Bt,v);
```
$[1, 2, 3]$

Ví dụ 2: Hãy biểu diễn p= x^2 trong cơ sở

B={1, (x-1) , $(x-1)^2$ } của $P_2[x]$

Ta tìm tọa độ của p trong B: bằng cách biểu diễn theo cơ sở chính tắc { 1, x , x^2 }

```
B:=transpose(matrix([[1,0,0],[-1,1,0],[1,-2,1]]));
```

$$B := \begin{bmatrix} 1 & -1 & 1 \\ 0 & 1 & -2 \\ 0 & 0 & 1 \end{bmatrix}$$

```
v:=vector([0,0,1]);
```
$v := [0, 0, 1]$

```
linsolve(B,v);
```
$[1, 2, 1]$

Suy ra : p = x^2 = 1 + 2(x-1) + $1(x-1)^2$

* Trường hợp, ta tìm tọa độ của nhiều vectơ v_1, v_2 , …, v_n trong cùng cơ sở B, ta giải hệ B^TX= V trong đó V là ma trận cột các vectơ v_1, v_2 , … , v_n và kết quả X là ma trận cột:

Ví dụ : Tìm tọa độ của v_1=(6,9,14) , v_2=(1,2,3) trong cơ sở

B={(1,1,1) , (1,1,2) , (1,2,3)}

```
Bt:=transpose(matrix([[1,1,1],[1,1,2],[1,2,3]]));
```

$$Bt := \begin{bmatrix} 1 & 1 & 1 \\ 1 & 1 & 2 \\ 1 & 2 & 3 \end{bmatrix}$$

```
V:=transpose(matrix([[6,9,14],[1,2,3]]));
```

$$V := \begin{bmatrix} 6 & 1 \\ 9 & 2 \\ 14 & 3 \end{bmatrix}$$

```
linsolve(Bt,V);
```

$$\begin{bmatrix} 1 & 0 \\ 2 & 0 \\ 3 & 1 \end{bmatrix}$$

□ Ma trận chuyển cơ sở :

Ma trận chuyển cơ sở từ B -> B1 là ma trận S là nghiệm : $B^TS = B1^T$

Ví dụ : B={(1,1,0) , (0,1,1) , (1,0,1) } và

B1={(0,0,1) , (1,-1,0) , (1,1,1) }

```
B:=transpose(matrix([[1,1,0],[0,1,1],[1,0,1]]));
```

$$B := \begin{bmatrix} 1 & 0 & 1 \\ 1 & 1 & 0 \\ 0 & 1 & 1 \end{bmatrix}$$

```
B1:=transpose(matrix([[0,0,1],[1,-1,0],[1,1,1]]));
```

$$B1 := \begin{bmatrix} 0 & 1 & 1 \\ 0 & -1 & 1 \\ 1 & 0 & 1 \end{bmatrix}$$

```
S=linsolve(B,B1);
```
$$S = \begin{bmatrix} \frac{-1}{2} & 0 & \frac{1}{2} \\ \frac{1}{2} & -1 & \frac{1}{2} \\ \frac{1}{2} & 1 & \frac{1}{2} \end{bmatrix}$$

```
S1=linsolve(B1,B);
```
$$S1 = \begin{bmatrix} -1 & \frac{1}{2} & \frac{1}{2} \\ 0 & \frac{-1}{2} & \frac{1}{2} \\ 1 & \frac{1}{2} & \frac{1}{2} \end{bmatrix}$$

IV. ÁNH XẠ TUYẾN TÍNH

1. NHÂN , ẢNH CỦA ÁNH XẠ TUYẾN TÍNH

Đối với các ánh xạ tuyến tính f trên R^n, ta tìm ảnh (Imf) , nhân (kerf) bằng cách dùng ma trận biểu diễn Af trong cơ sở chính tắc :

- Tìm ma trận hệ số A và Af = A^T
- Tìm ảnh f bằng lệnh: **rowspace(Af)**
- Tìm nhân f bằng lệnh: **kernel(A)**

Ví dụ : Cho ánh xạ tuyến tính f :$|R^3 \rightarrow |R^3$ xác định bởi :

f(x,y,z) = (x+2y-z , y+z , x+y-2z)

Tìm nhân và ảnh của f.

```
with(linalg):
A:=matrix([[1,2,-1],[0,1,1],[1,1,-2]]);
```
$$A := \begin{bmatrix} 1 & 2 & -1 \\ 0 & 1 & 1 \\ 1 & 1 & -2 \end{bmatrix}$$

Nhân của f là nghiệm của AX= 0 , nên:

```
kernel(A);
```
$\{[-3, 1, -1]\}$

Anh của f là không gian con sinh bởi các hàng của Af, nên :

```
Af:=transpose(A);
```
$$Af := \begin{bmatrix} 1 & 0 & 1 \\ 2 & 1 & 1 \\ -1 & 1 & -2 \end{bmatrix}$$

```
rowspace(Af);
```
$\{[0, 1, -1], [1, 0, 1]\}$

2. TÌM CÔNG THỨC CỦA ÁNH XẠ TUYẾN TÍNH KHI BIẾT ẢNH CỦA CƠ SỞ

Ví dụ : Cho ánh xạ tuyến tính f: $|R^2 \rightarrow |R^2$ thỏa: f(3,1)=(2,-4) ; f(1,1) = (0,2). Xác định f(x,y)

```
with(linalg):
```
Ma trận chuyển vị các vectơ cơ sở { (3,1) , (1,1) }

```
B:=transpose(matrix([[3,1],[1,1]]));
```
$$B := \begin{bmatrix} 3 & 1 \\ 1 & 1 \end{bmatrix}$$

Ma trận chuyển vị của ảnh B

```
ImB:=transpose(matrix([[2,-4],[0,2]]));
```
$$ImB := \begin{bmatrix} 2 & 0 \\ -4 & 2 \end{bmatrix}$$

Ma trận cột của vectơ bất kỳ v=(x,y)

```
v:=matrix(2,1,[x,y]);
```
$$v := \begin{bmatrix} x \\ y \end{bmatrix}$$

Ma trận tọa độ của v trong B

```
X:=linsolve(B,v);
```
$$X := \begin{bmatrix} \frac{1}{2}x - \frac{1}{2}y \\ -\frac{1}{2}x + \frac{3}{2}y \end{bmatrix}$$

```
f(x,y)=multiply(ImB,X);
```
$$f(x, y) = \begin{bmatrix} x - y \\ -3x + 5y \end{bmatrix}$$

$\Rightarrow$ f(x,y) = (x-y , -3x+5y)

3. TÌM MA TRẬN BIỂU DIỄN CỦA ÁNH XẠ TUYẾN TÍNH

<u>Ví dụ :</u>

Cho ánh xạ tuyến tính f: $|R^3 \to |R^2$ xác định bởi :

f(x,y,z) = (3x+2y-4z , x-5y+3z)

Tìm ma trận của f trong cơ sở B={(1,1,1) , (1,1,0) , (1,0,0) } đối với B'={(1,3) , (2,5) }

```
with(linalg):
```

Ma trận hệ số của f : f = AX

```
A:=matrix([[3,2,-4],[1,-5,3]]);
```
$$A := \begin{bmatrix} 3 & 2 & -4 \\ 1 & -5 & 3 \end{bmatrix}$$

Ma trận cột các vectơ cơ sở B

```
B:=transpose(matrix([[1,1,1],[1,1,0],[1,0,0]]));
```
$$B := \begin{bmatrix} 1 & 1 & 1 \\ 1 & 1 & 0 \\ 1 & 0 & 0 \end{bmatrix}$$

Ma trận ảnh của các vectơ trong cơ sở B là fB = AB

```
fB:=multiply(A,B);
```
$$fB := \begin{bmatrix} 1 & 5 & 3 \\ -1 & -4 & 1 \end{bmatrix}$$

Ma trận cột các vectơ cơ sở B'

```
BB:=transpose(matrix([[1,3],[2,5]]));
```
$$BB := \begin{bmatrix} 1 & 2 \\ 3 & 5 \end{bmatrix}$$

Ma trận của f trong B,B', giải hệ : (BB)Af = fB

```
f(B,BB)=linsolve(BB,fB);
```
$$f(B, BB) = \begin{bmatrix} -7 & -33 & -13 \\ 4 & 19 & 8 \end{bmatrix}$$

<u>Ví dụ</u> : Cho ánh xạ tuyến tính f: $|R^3 \to |R^3$ xác định bởi :

f(x,y,z) = (2y+z , x-4y , 3x)

Tìm ma trận của f trong :

a) cơ sở chính tắc của $|R^3$
b) cơ sở E={(1,1,1) , (1,1,0) , (1,0,0) }
c) Chứng tỏ hai ma trận của f trong hai cơ sở trên đồng dạng.

```
A:=matrix([[0,2,1],[1,-4,0],[3,0,0]]):
```

Ma trận hai cơ sở :

```
Eo:=transpose(matrix([[1,0,0],[0,1,0],[0,0,1]]));
E:=transpose(matrix([[1,1,1],[1,1,0],[1,0,0]]));
```

$$Eo := \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix} \quad E := \begin{bmatrix} 1 & 1 & 1 \\ 1 & 1 & 0 \\ 1 & 0 & 0 \end{bmatrix}$$

Ma trận ảnh của hai cơ sở :

```
fEo:=multiply(A,Eo);
fE:=multiply(A,E);
```

$$fEo := \begin{bmatrix} 0 & 2 & 1 \\ 1 & -4 & 0 \\ 3 & 0 & 0 \end{bmatrix} \quad fE := \begin{bmatrix} 3 & 2 & 0 \\ -3 & -3 & 1 \\ 3 & 3 & 3 \end{bmatrix}$$

Ma trận của f trong hai cơ sở:

```
f(Eo)=linsolve(Eo,fEo);
f(E)=linsolve(E,fE);
```

$$\mathrm{f}(Eo) = \begin{bmatrix} 0 & 2 & 1 \\ 1 & -4 & 0 \\ 3 & 0 & 0 \end{bmatrix} \quad \mathrm{f}(E) = \begin{bmatrix} 3 & 3 & 3 \\ -6 & -6 & -2 \\ 6 & 5 & -1 \end{bmatrix}$$

c) Ma trận chuyển cơ sở S từ Eo $\to$ E

```
S:=linsolve(Eo,E);
```

$$S := \begin{bmatrix} 1 & 1 & 1 \\ 1 & 1 & 0 \\ 1 & 0 & 0 \end{bmatrix}$$

Tính S^{-1}(fEo)S

```
multiply(1/S,fEo,S);
```

$$\begin{bmatrix} 3 & 3 & 3 \\ -6 & -6 & -2 \\ 6 & 5 & -1 \end{bmatrix}$$

$\Rightarrow$ fE = S^{-1}(fEo)S , nên hai ma trận nầy đồng dạng.

V. CHÉO HÓA MA TRẬN

1. TRỊ RIÊNG,VÉCTƠ RIÊNG

- *Đa thức đặc trưng của ma trận vuông A* : f(x) = det(x*I -A) , I là ma trận đơn vị
- *Trị riêng* : nghiệm của đa thức đặc trưng f(x)=det(x*I-A) = 0
- *Vectơ riêng ứng với trị riêng λ* : một nghiệm X≠O của hệ thuần nhất [λI-A]X= O
- *Không gian riêng ứng với trị riêng λ* : Không gian nghiệm của hệ [λI-A]X= O

Tên lệnh	Ý nghĩa
charpoly(A,x)	*Đa thức đặc trưng của A*
eigenvalues(A)	*Trị riêng của A*
eigenvectors(A)	*Trị riêng và Vectơ riêng của A được cho ở dạng danh sách: [λ, m, {v[1,i],... v[ni,i]}]* *trị riêng λ , bội m , vectơ riêng cơ sở.*
charmat(A,λ)	*Ma trận đặc trưng ứng với trị λ : $\lambda I - A$*
issimilar(A,B,'P')	*Kiểm tra xem A B đồng dạng hay không? và A=P^{-1} BP*

Ví dụ :

```
A := matrix(3,3, [1,-3,3,3,-5,3,6,-6,4]):
```

```
charpoly(A,x);
```
$x^3 - 12\,x - 16$

```
eigenvalues(A);
```
4, -2, -2

```
charmat(A,4);
```
$$\begin{bmatrix} 3 & 3 & -3 \\ -3 & 9 & -3 \\ -6 & 6 & 0 \end{bmatrix}$$

```
lst:=eigenvectors(A);
```
$lst := [4, 1, \{[1, 1, 2]\}], [-2, 2, \{[0, 1, 1], [1, 1, 0]\}]$

$[4, 1, \{[1, 1, 2]\}]$: trị riêng λ_1= 4 (bội 1) , vectơ cơ sở của không riêng (1,1,2)

$[-2, 2, \{[0, 1, 1], [1, 0, -1]\}]$: trị riêng λ_2= -2 (bội 2), vectơ cơ sở của không riêng (0,1,1) (1,0,-1)

Lấy danh sách vectơ riêng từ `lst`:

```
vr:=[lst[1,3,1],lst[2,3,1],lst[2,3,2]];
```
$vr := [[1, 1, 2], [1, 1, 0], [0, 1, 1]]$

```
P:=transpose(matrix(vr));
```
$$P := \begin{bmatrix} 1 & 1 & 0 \\ 1 & 1 & 1 \\ 2 & 0 & 1 \end{bmatrix}$$

```
multiply(1/P,A,P);
```
$$\begin{bmatrix} 4 & 0 & 0 \\ 0 & -2 & 0 \\ 0 & 0 & -2 \end{bmatrix}$$

Ví dụ:

```
A:=matrix(3,3, [1,-3,3,3,-5,3,6,-6,4]):
```

```
B:=diag(eigenvalues(A));
```
$$B := \begin{bmatrix} 4 & 0 & 0 \\ 0 & -2 & 0 \\ 0 & 0 & -2 \end{bmatrix}$$

```
issimilar(B,A,'P');
```
true

```
print(P);
```
$$\begin{bmatrix} 1 & -1 & 2 \\ 1 & -3 & 7 \\ 2 & -2 & 5 \end{bmatrix}$$

2. CHÉO HÓA MA TRẬN VUÔNG

- **Dạng Jordan**

  jordan(A)

  jordan(A, 'P')

trong đó : A là ma trận vuông , ‘P’ là ma trận chuyển đổi :

$P^{-1}AP = J$ (ma tận chéo)

Ví dụ:

```
with(linalg):
A:=matrix(3,3,[3,-2,0,-2,3,0,0,0,5]);
```
$$A := \begin{bmatrix} 3 & -2 & 0 \\ -2 & 3 & 0 \\ 0 & 0 & 5 \end{bmatrix}$$

```
jordan(A,P);
```

$$\begin{bmatrix} 1 & 0 & 0 \\ 0 & 5 & 0 \\ 0 & 0 & 5 \end{bmatrix}$$

```
print(P);
```

$$\begin{bmatrix} \frac{1}{2} & \frac{1}{2} & 0 \\ \frac{1}{2} & \frac{-1}{2} & 0 \\ 0 & 1 & 1 \end{bmatrix}$$

- **Chéo hóa trực giao**

```
with(linalg):
A:=matrix(3,3, [4,2,2,2,4,2,2,2,4]);
```

$$A := \begin{bmatrix} 4 & 2 & 2 \\ 2 & 4 & 2 \\ 2 & 2 & 4 \end{bmatrix}$$

```
lst:=eigenvectors(A);
```

$$lst := [8, 1, \{[1, 1, 1]\}], [2, 2, \{[1, 0, -1], [0, 1, -1]\}]$$

```
v:=[lst[1,3,1],lst[2,3,1],lst[2,3,2]];
```

$$v := [[1, 1, 1], [1, 0, -1], [0, 1, -1]]$$

```
V:=GramSchmidt(v);
```

$$V := \left[[1, 1, 1], [1, 0, -1], \left[\frac{-1}{2}, 1, \frac{-1}{2}\right] \right]$$

```
vt:=[normalize(V[1]),normalize(V[2]),
     normalize(V[3])];
```

$$vt := \left[\left[\frac{1}{3}\sqrt{3}, \frac{1}{3}\sqrt{3}, \frac{1}{3}\sqrt{3}\right], \left[\frac{1}{2}\sqrt{2}, 0, -\frac{1}{2}\sqrt{2}\right], \left[-\frac{1}{6}\sqrt{3}\sqrt{2}, \frac{1}{3}\sqrt{3}\sqrt{2}, -\frac{1}{6}\sqrt{3}\sqrt{2}\right] \right]$$

```
P:=transpose(matrix(vt));
```

$$P := \begin{bmatrix} \frac{1}{3}\sqrt{3} & \frac{1}{2}\sqrt{2} & -\frac{1}{6}\sqrt{3}\sqrt{2} \\ \frac{1}{3}\sqrt{3} & 0 & \frac{1}{3}\sqrt{3}\sqrt{2} \\ \frac{1}{3}\sqrt{3} & -\frac{1}{2}\sqrt{2} & -\frac{1}{6}\sqrt{3}\sqrt{2} \end{bmatrix}$$

```
orthog(P);
```

true

```
multiply(transpose(P),A,P);
```

$$\begin{bmatrix} 8 & 0 & 0 \\ 0 & 2 & 0 \\ 0 & 0 & 2 \end{bmatrix}$$

VI. DẠNG TOÀN PHƯƠNG

Để kiểm tra một ma trận A xác định dương hoặc âm, ta dùng lệnh : **definite(A, kind)**
trong đó:
A – là một ma trận vuông đối xứng.
kind – là 'positive_def', 'positive_semidef', 'negative_def', 'negative_semidef'

- Nếu các phần tử A là số thì cho kết quả true, false
- Nếu các phần tử A không là số thì cho kết quả là các điều kiện để true

Ví dụ 1:

```
A := matrix(2,2, [2,1,1,3]);
```

$$A := \begin{bmatrix} 2 & 1 \\ 1 & 3 \end{bmatrix}$$

```
definite(A, 'positive_def');
```

true

Ví dụ 2:

```
A := matrix(2,2, [a,b,b,d]);
```

$$A := \begin{bmatrix} a & b \\ b & d \end{bmatrix}$$

```
definite(A, 'positive_def');
```

$$-a < 0 \text{ and } -a\,d + b^2 < 0$$

Thực hành

1. Cho A= $\begin{bmatrix} 1 & 0 & 2 \\ 2 & -1 & 3 \\ 4 & 1 & 8 \end{bmatrix}$, B= $\begin{bmatrix} -3 & 2 \\ 0 & 1 \\ 7 & 4 \end{bmatrix}$. Tính

   A^{-1} ; AA^T ; B^TAB ; $(2A+BB^T)A^T$

2. Tính: $3A^2 - 2A + 5I$ với

   $$A = \begin{bmatrix} 1 & -2 & 3 \\ 2 & -4 & 1 \\ 3 & -5 & 2 \end{bmatrix}, \quad I = \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

3. Bằng phương pháp biến đổi sơ cấp, tìm ma trận nghịch đảo của các ma trận sau:

   a) $\begin{bmatrix} 7 & 3 & 2 \\ 9 & 4 & 3 \\ 5 & 3 & 1 \end{bmatrix}$ b) $\begin{bmatrix} 1 & 2 & 2 \\ 2 & -2 & 1 \\ 2 & 1 & -2 \end{bmatrix}$ c) $\begin{bmatrix} 1 & 1 & 1 & 1 \\ 1 & 1 & -1 & -1 \\ 1 & -1 & 1 & -1 \\ 1 & -1 & -1 & 1 \end{bmatrix}$

4. Tính định thức của ma trận :

   a) A= $\begin{bmatrix} x^2+1 & x & 0 & 0 \\ x & x^2+1 & x & 0 \\ 0 & x & x^2+1 & x \\ 0 & 0 & x & x^2+1 \end{bmatrix}$

   b) $\begin{bmatrix} x^2+1 & x & 0 & 0 & 0 \\ x & x^2+1 & x & 0 & 0 \\ 0 & x & x^2+1 & x & 0 \\ 0 & 0 & x & x^2+1 & x \\ 0 & 0 & 0 & x & x^2+1 \end{bmatrix}$

   c) Qui nạp định thức cấp n.

5. Giải các phương trình

a) $\begin{vmatrix} x & x+1 & x+2 \\ x+1 & x+3 & x+5 \\ x+4 & x+9 & x+14 \end{vmatrix} = 0$ b) $\begin{vmatrix} 1 & x & x^2 & x^3 \\ 1 & 2 & 4 & 8 \\ 1 & 3 & 9 & 27 \\ 1 & 4 & 16 & 64 \end{vmatrix} = 0$

6. Giải các hệ phương trình sau :

a) $\begin{cases} 2x_1 + 7x_2 + 3x_3 + x_4 = 6 \\ 3x_1 + 5x_2 + 2x_3 + 2x_4 = 4 \\ 9x_1 + 4x_2 + x_3 + 7x_4 = 2 \end{cases}$ *b)* $\begin{cases} 2x_1 + x_2 + x_3 - x_4 = 0 \\ 3x_1 - x_2 + 4x_3 + 3x_4 = 5 \\ 7x_1 + x_2 + 6x_3 + x_4 = 4 \end{cases}$

7. Tìm ma trận X sao cho :

a) $\begin{bmatrix} 1 & 2 \\ 1 & 3 \end{bmatrix} X = \begin{bmatrix} 3 & 6 & 9 \\ 4 & 8 & 12 \end{bmatrix}$ b) $\begin{bmatrix} 1 & 0 & 1 \\ 2 & -3 & 4 \\ 1 & -3 & 4 \end{bmatrix} X = \begin{bmatrix} 9 & -3 \\ 47 & -20 \\ 455 & -21 \end{bmatrix}$

8. Giải và biện luận các hệ sau:

a) $\begin{cases} x_1 + 2x_2 + 3x_3 + mx_4 = m + 2 \\ x_1 + x_2 + x_3 + mx_4 = m + 1 \\ 2x_1 + 3x_2 + 4x_3 + 2mx_4 = 2m + 3 \\ x_1 + x_2 + 2x_3 + 2mx_4 = m^2 + m + 2 \end{cases}$

b) $\begin{cases} mx + y + z = m \\ 2x + (m+1)y + (m+1)z = m - 1 \\ x + y + mz = 1 \end{cases}$

9. Tìm cơ sở và số chiều của không gian con sinh bởi M trong mỗi trường hợp sau :
   a) M={(1,2,3) , (1,0,1) , (2,2,4) , (2,4,6) } trong không gian $|R^3$
   b) M={v_1, v_2, v_3, v_4 , v_5} với v_1=(1,0,0,0), v_2=(1,1,1,1), v_3=(1,1,0,0) , v_4=(3,2,1,1), v_5=(3,3,2,2).

10. Cho U là không gian nghiệm của hệ phương trình : $\begin{cases} x_1 + 2x_2 + 3x_3 = 0 \\ 2x_1 + 3x_2 + 5x_3 = 0 \end{cases}$ và W là không gian nghiệm của phương trình $x_1+x_2+x_3 = 0$. Tìm cơ sở của các không gian con U, W, U∩W, U+W trong không gian $|R^3$.

11. Cho U=< (1,2,1) , (2,3,0) > , W=<(2,0,1) , (1,2,3)>
    Tìm cơ sở và số chiều của các không gian con U∩W và U+W trong không gian $|R^3$.

12. Tìm nhân và ảnh của ánh xạ tuyến tính :
    a). f :$|R^2 \to |R^2$ xác định f(x,y) = (2x-y,x-2y)
    b). f :$|R^3 \to |R^3$ xác định f(x,y,z)=(x+2y+2z ,2x-y-z ,-4x-3y-3z)

13. Cho ánh xạ tuyến tính f :$|R^2 \to |R^2$ xác định bởi :
    f(x,y) = (2x-3y,x+4y).
    Tìm ma trận của f trong cở sở B={(1,0) , (0,1)} đối với
    B'={(1,3) , (2,5) }

14. Cho ánh xạ tuyến tính $f : \mathbb{R}^3 \to \mathbb{R}^3$ xác định bởi :

$f(1,0,0)=(1,1,1)$; $f(-1,1,0)=(-2,-1,0)$; $f(0,-1,-1)=(2,1,0)$

a) Xác định f(x,y,z)

b) Tìm cơ sở , số chiều của Imf và kerf

c) Cho W={(x,y,z) | x+y+z=0}. Tìm cơ sở, số chiều của f(W)

15. Ma trận của ánh xạ tuyến tính f trong cơ sở

{(8,-6,7) , (-16,7,-13) , (9,-3,7)} là $\begin{bmatrix} 1 & -18 & 15 \\ -1 & -22 & 20 \\ 1 & -25 & 22 \end{bmatrix}$.

Tìm ma trận f trong cơ sở {(1,-2,1) , (3,-1,2) , (2,1,2)}

16. Tìm đa thức đặc trưng, trị riêng, vectơ riêng của ma trận:

$a)\ A = \begin{bmatrix} 2 & 2 & 1 \\ 1 & 3 & 1 \\ 1 & 2 & 2 \end{bmatrix}$

b) $B = \begin{bmatrix} 1 & -4 & -1 & -4 \\ 2 & 0 & 5 & -4 \\ -1 & 1 & -2 & 3 \\ -1 & 4 & -1 & 6 \end{bmatrix}$

17. Cho ma trận A_{nxn} định nghĩa : $a_{ij} = \begin{cases} 0 & \text{khi } i = j \\ 1 & \text{khi } i \neq j \end{cases}$

a) Tính det(A)

b) Tính đa thức đặc trưng của A

c) Xác định mọi trị riêng và vectơ cơ sở của không gian riêng tương ứng.

18. Chéo hóa các ma trận sau (nếu được)

$a)\ \begin{bmatrix} 1 & 0 & 0 \\ 2 & 2 & 0 \\ 3 & 3 & 3 \end{bmatrix}$ b) $\begin{bmatrix} 1 & -3 & 3 \\ 3 & -5 & 3 \\ 6 & -6 & 4 \end{bmatrix}$

19. Chéo hóa trực giao các ma trận (nếu được)

$a)\ \begin{bmatrix} 6 & -2 & 2 \\ -2 & 5 & 0 \\ 2 & 0 & 7 \end{bmatrix}$ b) $\begin{bmatrix} 1 & -3 & 1 \\ -3 & 1 & -1 \\ 1 & -1 & 5 \end{bmatrix}$

20. Dùng phép biến đổi trực giao đưa dạng toàn phương về dạng chính tắc :

a) $f(x,y) = 2x^2 + 8xy + 8y^2$

b) $f(x,y,z) = 3x^2 + 2y^2 + z^2 + 4xy + 4yz$

21. Phân loại đường cong bậc hai :

a) $3x^2 + 10xy + 3y^2 - 2x - 14y - 13 = 0$

b) $25x^2 - 14xy + 25y^2 + 64x - 64y - 224 = 0$